

# **MANUAL DE FUNÇÕES**

**EXCEL 2007**

**[M-T-003]**

**1ª edição**

**Documento reservado (interno)**

**Rio de Janeiro - RJ, 13 de outubro de 2011**

---

---

## Sumário

<b>FUNÇÕES DE DATA E HORA .....</b>	<b>1</b>
AGORA().....	1
ANO .....	1
MÊS.....	1
DIA .....	1
DIA.DA.SEMANA.....	1
DATA.....	1
HOJE().....	1
HORA.....	1
MINUTO .....	2
SEGUNDO .....	2
<b>FUNÇÕES MATEMÁTICAS E TRIGONOMÉTRICAS .....</b>	<b>2</b>
ABS .....	2
ALEATÓRIO .....	2
ARRED .....	2
ARREDONDAR.PARA.BAIXO .....	2
ARREDONDAR.PARA.CIMA .....	3
COMBIN .....	3
COS .....	4
FATORIAL .....	4
INT .....	4
MOD .....	4
MULT .....	5
PI .....	5
POTÊNCIA .....	5
RAIZ .....	5
ROMANO .....	5
SEN .....	6
SOMA .....	6
SOMAQUAD .....	6
SOMARPRODUTO .....	7
SOMASE .....	7
TAN .....	7
TRUNCAR .....	8
<b>FUNÇÕES ESTATÍSTICAS .....</b>	<b>8</b>
CONT.NÚM.....	8
CONT.SE .....	8
CONT.VALORES.....	9
CONTAR.VAZIO .....	9
DESVPAD.....	9
FREQÜÊNCIA .....	10
MAIOR.....	10
MÁXIMO .....	11
MED.....	11
MÉDIA .....	11
MENOR .....	11
MÍNIMO.....	12
MODO .....	12
ORDEM.....	12

---

<b>FUNÇÕES DE PROCURA E REFERÊNCIA.....</b>	<b>13</b>
COL.....	13
CORRESP.....	13
ESCOLHER.....	14
INFODADOSTABELADINÂMICA.....	14
LIN.....	15
PROCH.....	15
PROCV.....	16
TRANSPOR.....	17
<b>FUNÇÕES DE BANCO DE DADOS.....</b>	<b>18</b>
BDCONTAR.....	18
BDCONTARA.....	18
BDEXTRAIR.....	18
BDMÁX.....	19
BDMÉDIA.....	19
BDMÍN.....	19
BDMULTIPL.....	20
BDSOMA.....	20
<b>FUNÇÕES DE TEXTO.....</b>	<b>22</b>
ARRUMAR.....	22
CONCATENAR.....	22
DIREITA.....	23
ESQUERDA.....	23
EXATO.....	23
EXT.TEXTO.....	24
MAIÚSCULA.....	24
MINÚSCULA.....	24
MOEDA.....	24
MUDAR.....	25
NÚM.CARACT.....	25
PRI.MAIÚSCULA.....	26
PROCURAR.....	26
REPT.....	26
SUBSTITUIR.....	27
TEXTO.....	27
<b>FUNÇÕES LÓGICAS.....</b>	<b>28</b>
E.....	28
NÃO.....	28
OU.....	28
SE.....	29
<b>FUNÇÕES DE INFORMAÇÃO.....</b>	<b>30</b>
CÉL.....	30
FUNÇÕES É.....	32
INFORMAÇÃO.....	33
<b>FUNÇÕES FINANCEIRAS.....</b>	<b>34</b>
NPER.....	34
PGTO.....	34
TAXA.....	35
VF.....	36
VP.....	36

## FUNÇÕES DE DATA E HORA

### AGORA()

Retorna o número de série seqüencial da data e hora atuais. Se o formato da célula era **Geral** antes de a função ser inserida, o resultado será formatado como uma data.

	A
1	<b>Data</b>
2	5/7/2008
3	5/7/2010

### ANO

Retorna o ano correspondente a uma data. O ano é retornado como um inteiro no intervalo de 1900-9999.

Exemplo: =ANO(A2) retorna 2008

	A
1	<b>Data</b>
2	5/7/2008
3	5/7/2010

### MÊS

Retorna o mês de uma data representado por um número de série. O mês é fornecido como um inteiro, variando de 1 (janeiro) a 12 (dezembro).

Exemplo: =MÊS(A2) retorna 7.

	A
1	<b>Data</b>
2	5/7/2008
3	5/7/2010

### DIA

Retorna o dia de uma data representado por um número de série. O dia é dado como um inteiro que varia de 1 a 31.

Exemplo: =DIA(A2) retorna 5

	A
1	<b>Data</b>
2	5/7/2008
3	5/7/2010

### DIA.DA.SEMANA

Retorna o dia da semana correspondente a uma data. O dia é dado como um inteiro, variando de 1 (domingo) a 7 (sábado), por padrão.

As datas devem ser inseridas com a função DATA, como resultado de outras fórmulas ou funções ou digitadas em uma célula.

Exemplo: =DIA.DA.SEMANA(A2) retorna 7, ou seja, Domingo.

### DATA

Retorna o número de série seqüencial que representa uma determinada data. Se o formato da célula era **Geral** antes de a função ser inserida, o resultado será formatado como uma data.

#### Sintaxe

**DATA(ano;dia;mês)**

Exemplo: =DATA(2006;7;24) retorna 24/07/2006, se a célula estiver formatada como Geral.

### HOJE()

Retorna a data atual numa célula. Se o formato da célula era **Geral** antes de a função ser inserida, o resultado será formatado como uma data.

#### Comentários:

O Excel armazena datas como números de série seqüenciais para que eles possam ser usados em cálculos. Por padrão, 1° de janeiro de 1900 é o número de série 1 e 1° de janeiro de 2008 é o número de série 39448 porque está 39.448 dias após 1° de janeiro de 1900.

	A
1	14:05:30
2	06:10:45

### HORA

Retorna a hora de um valor de tempo.

Exemplo: =HORA(A1) retorna 14; =HORA(A2) retorna 6

	A
1	14:05:30
2	06:10:45

### MINUTO

Retorna os minutos de um valor de tempo.

Exemplo: =MINUTO(A1) retorna 5; =MINUTO(A2) retorna 10.

	A
1	14:05:30
2	06:10:45

### SEGUNDO

Retorna os segundos de um valor de tempo.

Exemplo: =SEGUNDO(A1) retorna 30; =SEGUNDO(A2) retorna 45.

## FUNÇÕES MATEMÁTICAS E TRIGONOMÉTRICAS

### ABS

Retorna o valor absoluto de um número. O valor absoluto de um número é o próprio número sem o respectivo sinal.

Exemplos: =ABS(2) retorna o valor absoluto de 2 (2); =ABS(-2) retorna o valor absoluto de -2 (2).

### ALEATÓRIO

Retorna um número aleatório maior ou igual a 0 e menor que 1 distribuído igualmente. Um novo número aleatório é retornado toda vez que a planilha for calculada.

Exemplo: =ALEATÓRIO()

- Para gerar um número real aleatório entre 0 e 50, por exemplo, use: =ALEATÓRIO()\*50. Se quiser somente mostrar a parte inteira desse número aleatório use a função INT, juntamente com a função ALEATÓRIO, da seguinte maneira: =INT(ALEATÓRIO()\*50)

### ARRED

Arredonda um número até uma quantidade especificada de dígitos.

#### Sintaxe

**ARRED(núm;núm\_dígitos)**

Núm é o número que você deseja arredondar.

Núm\_dígitos especifica o número de dígitos para o qual você deseja arredondar núm.

#### Comentários

- Se núm\_dígitos for maior que 0, então núm será arredondado para o número especificado de casas decimais.
- Se núm\_dígitos for 0, então núm será arredondado para o inteiro mais próximo.
- Se núm\_dígitos for menor que 0, então núm será arredondado para a esquerda da vírgula decimal.

Fórmula	Descrição (resultado)
=ARRED(2,15; 1)	Arredonda 2,15 para uma casa decimal (2,2)
=ARRED(2,149; 1)	Arredonda 2,149 para uma casa decimal (2,1)
=ARRED(-1,475; 2)	Arredonda -1,475 para duas casas decimais (-1,48)
=ARRED(21,5; -1)	Arredonda 21,5 para uma casa à esquerda da vírgula decimal (20)

### ARREDONDAR.PARA.BAIXO

Arredonda um número para baixo até zero.

#### Sintaxe

**ARREDONDAR.PARA.BAIXO(núm;núm\_dígitos)**

Núm é qualquer número real que se deseja arredondar.

Núm\_dígitos é o número de dígitos para o qual se deseja arredondar núm.

### Comentários

- ARREDONDAR.PARA.BAIXO funciona como ARRED, com a diferença de sempre arredondar um número para baixo.
- Se núm\_dígitos for maior do que 0, então o número será arredondado para baixo pelo número de casas decimais especificado.
- Se núm\_dígitos for 0, núm será arredondado para baixo até o inteiro mais próximo.
- Se núm\_dígitos for menor do que 0, então o número será arredondado para baixo à esquerda da vírgula decimal.

Fórmula	Descrição (resultado)
=ARREDONDAR.PARA.BAIXO(3,2; 0)	Arredonda 3,2 para baixo, para zero casa decimal (3)
=ARREDONDAR.PARA.BAIXO(76,9;0)	Arredonda 76,9 para baixo, para zero casa decimal (76)
=ARREDONDAR.PARA.BAIXO(3,14159; 3)	Arredonda 3,14159 para baixo, para três casas decimais (3,141)
=ARREDONDAR.PARA.BAIXO(-3,14159; 1)	Arredonda -3,14159 para baixo, para uma casa decimal (-3,1)
=ARREDONDAR.PARA.BAIXO(31415,92654; -2)	Arredonda 31415,92654 para baixo, para 2 casas à esquerda da vírgula decimal (31400)

### ARREDONDAR.PARA.CIMA

Arredonda um número para cima afastando-o de zero.

#### Sintaxe

**ARREDONDAR.PARA.CIMA(núm;núm\_dígitos)**

Núm é qualquer número real que se deseja arredondar.

Núm\_dígitos é o número de dígitos para o qual se deseja arredondar núm.

### Comentários

- ARREDONDAR.PARA.CIMA funciona como ARRED, com a diferença de sempre arredondar um número para cima.
- Se núm\_dígitos for maior do que 0, então o número será arredondado para cima pelo número de casas decimais especificado.
- Se núm\_dígitos for 0, núm será arredondado para cima até o inteiro mais próximo.
- Se núm\_dígitos for menor do que 0, então o número será arredondado para cima à esquerda da vírgula decimal.

Fórmula	Descrição (resultado)
=ARREDONDAR.PARA.CIMA(3,2;0)	Arredonda 3,2 para cima, para zero casa decimal (4)
=ARREDONDAR.PARA.CIMA(76,9;0)	Arredonda 76,9 para cima, para zero casa decimal (77)
=ARREDONDAR.PARA.CIMA(3,14159; 3)	Arredonda 3,14159 para cima, para três casas decimais (3,142)
=ARREDONDAR.PARA.CIMA(-3,14159; 1)	Arredonda -3,14159 para cima, para uma casa decimal (-3,2)
=ARREDONDAR.PARA.CIMA(31415,92654; -2)	Arredonda 31415,92654 para cima, para 2 casas à esquerda da vírgula decimal (31500)

### COMBIN

Retorna o número de combinações de um determinado número de itens. Use COMBIN para determinar o número total possível de grupos para determinado número de objetos.

#### Sintaxe

**COMBIN(núm;núm\_escolhido)**

Núm é o número de itens.

Núm\_escolhido é o número de itens em cada combinação.

Exemplo: =COMBIN(8;2) retorna as possíveis combinações de equipes de duas pessoas que podem ser formadas de 8 candidatos (28).

## COS

Retorna o cosseno do ângulo dado.

### Sintaxe

**COS(núm)**

Núm é o ângulo em radianos para o qual você deseja obter o cosseno.

### Comentários

Se o ângulo estiver em graus, multiplique-o por PI()/180 ou use a função COS para convertê-lo em radianos.

Fórmula	Descrição (resultado)
=COS(1,047)	O cosseno de 1,047 radianos (0,500171)
=COS(60*PI()/180)	O cosseno de 60 graus (0,5)
=COS(RADIANS(60))	O cosseno de 60 graus (0,5)

## FATORIAL

Retorna o fatorial de um número. O fatorial de um número é igual ao número 1\*2\*3\*...\*

### Sintaxe

**FATORIAL(núm)**

Núm é o número não-negativo do qual você deseja obter o fatorial. Se núm não for um inteiro, será truncado.

Fórmula	Descrição (resultado)
=FATORIAL(5)	O fatorial de 5 ou 1*2*3*4*5 (120)
=FATORIAL(1,9)	O fatorial do número inteiro de 1,9 (1)
=FATORIAL(0)	O fatorial de 0 (1)
=FATORIAL(-1)	Os números negativos causam um valor de erro (#NÚM!)
=FATORIAL(1)	O fatorial de 1 (1)

## INT

Arredonda um número para baixo até o número inteiro mais próximo.

### Sintaxe

**INT(núm)**

Núm é o número real que se deseja arredondar para baixo até um inteiro.

Exemplos:		Fórmula	Descrição (resultado)
1	Dados	=INT(8,9)	Arredonda 8,9 para baixo (8)
2	19,5	=INT(-8,9)	Arredonda -8,9 para baixo (-9)
		=A2-INT(A2)	Retorna a parte decimal de um número real positivo na célula A2 (0,5)

## MOD

Retorna o resto depois da divisão de núm por divisor. O resultado possui o mesmo sinal que divisor.

### Sintaxe

**MOD(núm,divisor)**

Núm é o número para o qual você deseja encontrar o resto.

Divisor é o número pelo qual você deseja dividir o número.

Fórmula	Descrição (resultado)
=MOD(3; 2)	O resto de 3/2 (1)

=MOD(-3; 2)	O resto de -3/2. O sinal é igual ao do divisor (1)
=MOD(3; -2)	O resto de 3/-2. O sinal é igual ao do divisor (-1)
=MOD(-3; -2)	O resto de -3/-2. O sinal é igual ao do divisor (-1)

## MULT

	A
1	Dados
2	5
3	15
4	30

Multiplica todos os números fornecidos como argumentos e retorna o produto.

### Sintaxe

**MULT(núm1;núm2;...)**

Núm1, núm2,... são números de 1 a 30 que você deseja multiplicar.

Fórmula	Descrição (resultado)
=MULT(A2:A4)	Multiplica os números acima (2250)
=MULT(A2:A4; 2)	Multiplica os números acima e 2 (4500)

## PI

	A
1	Raio
2	3

Retorna o número 3,14159265358979, a constante matemática PI, com precisão de até 15 dígitos.

Fórmula	Descrição (resultado)
=PI()	O PI (3,14159265358979)
=PI()/2	O PI/2 (1,570796327)
=PI()*(A2^2)	A área de um círculo com o raio indicado na figura ao lado (28,27433388)

## POTÊNCIA

Fornece o resultado de um número elevado a uma potência.

### Sintaxe

**POTÊNCIA(núm;potência)**

Núm é o número base. Pode ser qualquer número real.

Potência é o expoente para o qual a base é elevada.

### Comentários

O operador "^" pode substituir POTÊNCIA para indicar a potência pela qual o número base deve ser elevado, tal como em 5^2.

Fórmula	Descrição (resultado)
=POTÊNCIA(5;2)	5 ao quadrado (25)
=POTÊNCIA(98,6;3,2)	98,6 elevado à potência 3,2 (2401077)
=POTÊNCIA(4;5/4)	4 elevado à potência 5/4 (5,656854)

## RAIZ

Retorna uma raiz quadrada positiva.

### Sintaxe

**RAIZ(núm)**

Núm é o número do qual você deseja obter a raiz quadrada.

Se núm for negativo, RAIZ retornará o valor de erro #NÚM!.

Exemplos: =RAIZ(16) retorna 4.

## ROMANO

Converte um algarismo arábico em romano, como texto.

### Sintaxe

**ROMANO(núm;forma)**

Núm é o algarismo arábico a ser convertido.

Forma é o algarismo que especifica o tipo de algarismo romano desejado. O estilo do algarismo romano varia de clássico a simplificado, tornando-se mais conciso à medida que o valor da forma aumenta. Consulte o exemplo de ROMANO(499;0) seguinte.

Forma	Tipo
0 ou omitido	Clássico
1	Mais conciso. Consulte o exemplo abaixo
2	Mais conciso. Consulte o exemplo abaixo
3	Mais conciso. Consulte o exemplo abaixo
4	Simplificado
VERDADEIRO	Clássico
FALSO	Simplificado

Se o número for maior do que 3999, o valor de erro #VALOR! Será retornado.

Fórmula	Descrição (resultado)
=ROMANO(499;0)	Estilo de algarismo romano clássico para 499 (CDXCIX)
=ROMANO(499;1)	Versão mais concisa para 499 (LDVLIIV)
=ROMANO(499;2)	Versão mais concisa para 499 (XDIX)
=ROMANO(499;3)	Versão mais concisa para 499 (VDIV)
=ROMANO(499;4)	Versão mais concisa para 499 (ID)
=ROMANO(2013;0)	Estilo de algarismo romano clássico para 2013 (MMXIII)

## SEN

Retorna o seno de um ângulo dado.

### Sintaxe

#### SEN(núm)

Núm é o ângulo em radianos para o qual você deseja obter o seno.

### Comentários

Se o argumento estiver em graus, multiplique-o por PI()/180 ou use a função RADIANOS para convertê-lo em radianos.

Fórmula	Descrição (resultado)
=SEN(PI())	O seno de radianos de pi (0, aproximadamente)
=SEN(PI()/2)	O seno de radianos de pi/2 (1)
=SEN(30*PI()/180)	O seno de 30 graus (0,5)
=SEN(RADIANOS(30))	O seno de 30 graus (0,5)

	A
1	Dados
2	-5
3	15
4	30
5	5
6	VERDADEIRO

## SOMA

Retorna a soma de todos os números na lista de argumentos.

### Sintaxe

#### SOMA(núm1;núm2; ...)

Núm1, núm2,... são argumentos de 1 a 30 que se deseja somar.

Fórmula	Descrição (resultado)
=SOMA(3; 2)	Adiciona 3 e 2 (5)
=SOMA("5"; 15; VERDADEIRO)	Adiciona 5, 15 e 1, porque os valores de texto são traduzidos em números e o valor lógico VERDADEIRO é traduzido no número 1 (21)
=SOMA(A2:A4)	Adiciona os três primeiros números da coluna acima (40)
=SOMA(A2:A4; 15)	Adiciona os três primeiros números da coluna acima e 15 (55)
=SOMA(A5;A6; 2)	Adiciona os valores das duas últimas linhas acima e 2. Como valores não-numéricos em referências não são traduzidos, os valores na coluna acima são ignorados (2)

## SOMAQUAD

Retorna a soma dos quadrados dos argumentos.

### Sintaxe

#### SOMAQUAD(núm1;núm2; ...)

Núm1, núm2,... são argumentos de 1 a 30 para os quais se deseja a soma dos quadrados.

Fórmula	Descrição (resultado)
=SOMAQUAD(3; 4)	A soma dos quadrados de 3 e 4 (25)

## SOMARPRODUTO

Multiplica os componentes correspondentes nas matrizes fornecidas e retorna a soma destes produtos.

### Sintaxe

**SOMARPRODUTO(matriz1;matriz2;matriz3; ...)**

Matriz1, matriz2, matriz3,... são matrizes de 2 a 30 cujos componentes se deseja multiplicar e depois somar.

### Comentários

- Os argumentos da matriz devem ter a mesma dimensão. Se não tiverem, SOMARPRODUTO fornecerá o valor de erro #VALOR!.

	A	B	C	D
1	Matriz 1	Matriz 1	Matriz 2	Matriz 2
2	3	4	2	7
3	8	6	6	7
4	1	9	5	3

Exemplo:

Fórmula	Descrição (resultado)
=SOMARPRODUTO(A2:B4; C2:D4)	Multiplica todos os componentes das duas matrizes e depois adiciona os produtos — ou seja, $3*2 + 4*7 + 8*6 + 6*7 + 1*5 + 9*3$ (156)

### Comentários

O exemplo anterior retorna o mesmo resultado que a fórmula  $SOMA(A2:B4*C2:D4)$  inserida como uma matriz. Usar matrizes oferece uma solução mais geral para fazer operações semelhantes a SOMARPRODUTO. Por exemplo, você pode calcular a soma dos quadrados dos elementos em A2:B4 usando a fórmula  $SOMA(A1:B4^2)$  e pressionando CTRL+SHIFT+ENTER.

## SOMASE

Adiciona as células especificadas por um determinado critério.

### Sintaxe

**SOMASE(intervalo;critérios;intervalo\_soma)**

Intervalo é o intervalo de células que se deseja calcular.

Critérios são os critérios na forma de um número, expressão ou texto, que define quais células serão adicionadas. Por exemplo, os critérios podem ser expressos como 32, "32", ">32", "maçãs".

Intervalo\_soma são as células que serão realmente somadas.

	A	B
1	Valor de propriedade	Comissão
2	100.000	7.000
3	200.000	14.000
4	300.000	21.000
5	400.000	28.000

=**SOMASE(A2:A5;">160000";B2:B5)** retorna a soma das comissões para valores de propriedade acima de 160000 (63.000).

## TAN

Retorna a tangente de um determinado ângulo.

### Sintaxe

**TAN(núm)**

Núm é o ângulo em radianos para o qual se deseja a tangente.

### Comentários

Se o argumento estiver em graus, multiplique-o por  $PI()/180$  ou use a função RADIANOS para convertê-lo em radianos.

Fórmula	Descrição (resultado)
=TAN(0,785)	A tangente de 0,785 radianos (0,99920)
=TAN(45*PI()/180)	A tangente de 45 graus (1)
=TAN(RADIANOS(45))	A tangente de 45 graus (1)

## TRUNCAR

Trunca um número para um inteiro removendo a parte fracionária do número.

### Sintaxe

**TRUNCAR(núm;núm\_dígitos)**

Núm é o número que se deseja truncar.

Núm\_dígitos é um número que especifica a precisão da operação. O valor padrão para núm\_dígitos é 0 (zero).

### Comentários

TRUNCAR e INT são semelhantes pois os dois retornam inteiros. TRUNCAR remove a parte fracionária do número. INT arredonda para menos até o número inteiro mais próximo de acordo com o valor da parte fracionária do número. INT e TRUNCAR são diferentes apenas quando usam números negativos: TRUNCAR(-4,3) retorna -4, mas INT(-4,3) retorna -5, porque -5 é o número menor.

Fórmula	Descrição (resultado)
=TRUNCAR(8,9)	A parte inteira de 8,9 (8)
=TRUNCAR(-8,9)	A parte inteira de -8,9 (-8)
=TRUNCAR(PI())	A parte inteira de pi (3)

## FUNÇÕES ESTATÍSTICAS

### CONT.NÚM

Conta quantas células contêm números e também os números na lista de argumentos.

### Sintaxe

**CONT.NÚM(valor1;valor2;...)**

Valor1; valor2, ... são argumentos de 1 a 30 que contêm ou se referem a uma variedade de diferentes tipos de dados, mas somente os números são contados.

A		Fórmula	Descrição (resultado)
1	Dados		
2	Vendas		
3	8/12/2008	=CONT.NÚM(A2:A8)	Conta o número de células que contêm números na lista ao lado (3)
4			
5	19	=CONT.NÚM(A5:A8)	Conta o número de células que contêm números nas últimas 4 linhas da lista (2)
6	22,24		
7	VERDADEIRO		
8	#DIV/0!	=CONT.NÚM(A2:A8;2)	Conta o número de células que contêm números na lista e o valor 2 (4)

### CONT.SE

Calcula o número de células não vazias em um intervalo que corresponde a determinados critérios.

### Sintaxe

**CONT.SE(intervalo;critérios)**

Intervalo é o intervalo de células no qual se deseja contar células não vazias.

Critérios é o critério na forma de um número, expressão ou texto que define quais células serão contadas.

Por exemplo, os critérios podem ser expressos como 32, "32", ">32", "maçãs".

### Comentários

O Microsoft Excel fornece funções adicionais que podem ser usadas para analisar seus dados com base em uma condição. Por exemplo, para calcular uma soma baseada em uma seqüência de caracteres de texto ou em um número contido em um intervalo, use a função de planilha SOMASE. Para que a fórmula retorne um de dois valores com base em uma condição, como uma bonificação de vendas baseada em um valor de vendas especificado, use a função de planilha SE.

A		B		Fórmula	Descrição (resultado)
1	Dados	Dados			
2	maçãs	32			
3	laranjas	54			
4	pêssegos	75			
5	maçãs	86			

  

Fórmula	Descrição (resultado)
=CONT.SE(A2:A5;"maçãs")	O número de células com maçãs na primeira coluna (2)
=CONT.SE(B2:B5;">55")	O número de células com um valor maior que 55 na segunda coluna (2)

## CONT.VALORES

Calcula o número de células não vazias e os valores na lista de argumentos. Use CONT.VALORES para calcular o número de células com dados em um intervalo ou matriz.

### Sintaxe

**CONT.VALORES(valor1;valor2;...)**

Valor1; valor2;... são argumentos de 1 a 30 que representam os valores que você deseja calcular. Neste caso, um valor é qualquer tipo de informações, incluindo texto vazio (""), mas não incluindo células em branco.

	A	Fórmula	Descrição (resultado)
1	Dados		
2	Vendas		
3	8/12/2008	=CONT.VALORES(A2:A8)	Conta o número de células não vazias na lista (6)
4			
5	19	=CONT.VALORES(A5:A8)	Conta o número de células não vazias nas últimas 4 linhas da lista (4)
6	22,24		
7	VERDADEIRO	=CONT.VALORES(A1:A7;2)	Conta o número de células não vazias na lista e o valor 2 (7)
8	#DIV/0!	=CONT.VALORES(A1:A7;"Dois")	Conta o número de células não vazias na lista e o valor "Dois" (7)

## CONTAR.VAZIO

Conta o número de células vazias no intervalo especificado.

### Sintaxe

**CONTAR.VAZIO(intervalo)**

Intervalo é o intervalo no qual se deseja contar as células em branco.

### Comentários

Células com fórmulas que retornam "" (texto vazio) também são contadas. Células com valores nulos não são contadas.

	A	B	Fórmula	Descrição (resultado)
1	Dados	Dados		
2				
3	6	=SE(B4<30;"",B4)	=CONTAR.VAZIO(A2:B5)	Conta as células vazias no intervalo. A fórmula retorna texto vazio. (4)
4		27		
5	4	34		

## DESVPAD

Calcula o desvio padrão a partir de uma amostra. O desvio padrão é uma medida do grau de dispersão dos valores em relação ao valor médio (a média).

### Sintaxe

**DESVPAD(núm1;núm2;...)**

Núm1;núm2;... são de 1 a 30 argumentos numéricos correspondentes a uma amostra de população. Você também pode usar uma única matriz ou uma referência a uma matriz em vez de argumentos separados por pontos-e-vírgulas.

### Exemplo

Suponha que 10 ferramentas feitas na mesma máquina durante a produção são coletadas como uma amostra aleatória e avaliadas em termos de resistência à ruptura.

	A	Fórmula	Descrição (resultado)
1	Intensidade		
2	1345		
3	1301	=DESVPAD(A2:A11)	O desvio padrão de resistência à ruptura (27,46391572)
4	1368		
5	1322		
6	1310		
7	1370		
8	1318		
9	1350		
10	1303		
11	1299		

## FREQUÊNCIA

Calcula a frequência com que os valores ocorrem em um intervalo de valores e, em seguida, retorna uma matriz vertical de números. Por exemplo, use FREQUÊNCIA para contar o número de resultados de teste. Pelo fato de FREQUÊNCIA retornar uma matriz, deve ser inserida como uma fórmula matricial.

### Sintaxe

**FREQUÊNCIA(matriz\_dados;matriz\_bin)**

Matriz\_dados é uma matriz ou uma referência a um conjunto de valores cujas frequências você deseja contar. Se matriz\_dados não contiver valores, FREQUÊNCIA retornará uma matriz de zeros.

Matriz\_bin é uma matriz ou referência a intervalos nos quais você deseja agrupar os valores contidos em matriz\_dados. Se matriz\_bin não contiver valores, FREQUÊNCIA retornará o número de elementos em matriz\_dados.

### Comentários

- FREQUÊNCIA é inserida como uma fórmula matricial depois de selecionado um intervalo de células adjacentes no qual você deseja que a distribuição fornecida apareça.
- O número de elementos da matriz retornada é um a mais do que o número de elementos de matriz\_bin. O elemento adicional da matriz retornada apresenta a contagem dos valores que estiverem acima do intervalo mais alto. Por exemplo, ao contar três intervalos de valores inseridos em três células, certifique-se de inserir a FREQUÊNCIA em quatro células para os resultados. A célula adicional retornará o número de valores de matriz\_dados que forem maiores do que o valor do terceiro intervalo.
- FREQUÊNCIA ignora células em branco e texto.
- As fórmulas que fornecem matrizes devem ser inseridas como fórmulas matriciais.

	A	B	Fórmula	Descrição (resultado)
1	Resultados	Intervalos		
2	79	70	=FREQUÊNCIA(A2:A10;B2:B5)	O número de resultados menores ou iguais a 70 (1)
3	85	79		
4	78	89		
5	85			
6	50			O número de resultados no intervalo 71-79 (2)
7	81			O número de resultados no intervalo 80-89 (4)
8	95			O número de resultados maiores ou iguais a 90 (2)
9	88			
10	97			

**Observação** A fórmula no exemplo precisa ser inserida como uma fórmula de matriz. Após copiar o exemplo em uma planilha em branco, selecione o intervalo A13:A16 iniciando pela célula da fórmula. Pressione F2 e, em seguida, pressione CTRL+SHIFT+ENTER. Se não for inserida como uma fórmula de matriz, a fórmula terá como único resultado 1.

## MAIOR

Retorna o maior valor k-ésimo de um conjunto de dados. Você pode usar esta função para selecionar um valor de acordo com a sua posição relativa. Por exemplo, você pode usar MAIOR para obter o primeiro, o segundo e o terceiro resultados.

### Sintaxe

**MAIOR(matriz;k)**

Matriz é a matriz ou intervalo de dados cujo maior valor k-ésimo você deseja determinar.

K é a posição (do maior) na matriz ou intervalo de célula de dados a ser fornecida.

	A	B	Fórmula	Descrição (resultado)
1	Dados	Dados		
2	3	4	=MAIOR(A2:B6;3)	O terceiro maior entre os números (5)
3	5	2		
4	3	4	=MAIOR(A2:B6;7)	O sétimo maior entre os números (4)
5	5	6		
6	4	7		

## MÁXIMO

Retorna o valor máximo de um conjunto de valores.

### Sintaxe

**MÁXIMO**(núm1;núm2; ...)

Núm1, núm2,... são de 1 a 30 números para os quais você deseja saber o valor máximo.

	A	Fórmula	Descrição (resultado)
1	Dados		
2	10	=MÁXIMO(A2:A6)	O maior entre os números (27)
3	7		
4	9	=MÁXIMO(A2:A6; 30)	O maior entre os números e 30 (30)
5	27		
6	2		

## MED

Retorna a mediana dos números indicados. A mediana é o número no centro de um conjunto de números; isto é, metade dos números possui valores que são maiores do que a mediana e a outra metade possui valores menores.

### Sintaxe

**MED**(núm1;núm2;...)

Núm1; núm2;... são de 1 a 30 números dos quais você deseja obter a mediana.

Se houver uma quantidade par de números no conjunto, MED calculará a média dos dois números do meio.

	A	Fórmula	Descrição (resultado)
1	Dados		
2	1		
3	2		
4	3		
5	4	=MED(A2:A6)	A mediana dos primeiros 5 números da lista acima (3)
6	5		
7	6	=MED(A2:A7)	A mediana de todos os números acima ou a média de 3 e 4 (3,5)

## MÉDIA

Retorna a média aritmética dos argumentos.

### Sintaxe

**MÉDIA**(núm1;núm2; ...)

Núm1; núm2;... são de 1 a 30 argumentos numéricos para os quais você deseja obter a média.

	A	Fórmula	Descrição (resultado)
1	Dados		
2	10		
3	7		
4	9	=MÉDIA(A2:A6)	A média dos números acima (11)
5	27		
6	2	=MÉDIA(A2:A6; 5)	A média dos números acima e 5 (10)

## MENOR

Retorna o menor valor k-ésimo do conjunto de dados. Use esta função para retornar valores com uma posição específica relativa em um conjunto de dados.

### Sintaxe

**MENOR**(matriz;k)

Matriz é uma matriz ou intervalo de dados numéricos cujo menor valor k-ésimo você deseja determinar.

K é a posição (a partir do menor) na matriz ou intervalo de dados a ser fornecido.

	A	B	Fórmula	Descrição (resultado)
1	Dados	Dados		
2	3	1		
3	4	4		
4	5	8		
5	2	3		
6	3	7		
7	4	12		
8	6	54		
9	4	8		
10	7	23		
			=MENOR(A2:A10;4)	O quarto menor número na primeira coluna (4)
			=MENOR(B2:B10;2)	O segundo menor número na segunda coluna (3)

## MÍNIMO

Retorna o menor número na lista de argumentos.

Sintaxe

**MÍNIMO**(núm1;núm2;...)

Núm1, núm2,... são de 1 a 30 números dos quais você deseja saber o valor mínimo.

	A
1	Dados
2	10
3	7
4	9
5	27
6	2

Fórmula	Descrição (resultado)
=MÍNIMO(A2:A6)	O menor entre os números (2)
=MÍNIMO(A2:A6;0)	O menor entre os números ao lado e 0 (0)

## MODA

Retorna o valor que ocorre com mais freqüência em uma matriz ou intervalo de dados. Assim como MED, MODA é uma medida de local.

Sintaxe

**MODA**(núm1;núm2;...)

Núm1, núm2,... são argumentos de 1 a 30 para os quais você deseja calcular o modo. Você também pode usar uma única matriz ou referência a uma matriz em vez de argumentos separados por pontos-e-vírgulas.

	A
1	Dados
2	5,6
3	4
4	4
5	3
6	2
7	4

Fórmula	Descrição (resultado)
=MODA(A2:A7)	O modo ou o número que ocorre com mais freqüência (4)

## ORDEM

Retorna a posição de um número em uma lista de números. A ordem de um número é seu tamanho em relação a outros valores de uma lista. (Se você fosse classificar a lista, a ordem do número seria a sua posição).

Sintaxe

**ORDEM**(núm;ref;ordem)

Núm é o número cuja posição se deseja encontrar.

Ref é uma matriz ou referência a uma lista de números. Valores não numéricos em ref são ignorados.

Order é um número que especifica como posicionar um número em uma ordem.

- Se ordem for 0 ou omitido, o Microsoft Excel ordenará o número como se ref fosse uma lista classificada na ordem descendente.
- Se ordem for qualquer valor diferente de zero, o Microsoft Excel ordenará o número como se ref fosse uma lista classificada na ordem ascendente.

### Comentários

- ORDEM fornece números duplicados na mesma ordem. No entanto, a presença de números duplicados afeta a ordem dos números subsequentes. Por exemplo, em uma lista de números inteiros classificados em ordem crescente, se o número 10 aparecer duas vezes e tiver uma ordem de 5, então 11 teria uma ordem de 7 (nenhum número teria uma ordem de 6).

	A
1	Dados
2	7
3	3,5
4	3,5
5	1
6	2

Fórmula	Descrição (resultado)
=ORDEM(A3;A2:A6;1)	A ordem de 3,5 na lista acima (3)
=ORDEM(A2;A2:A6;1)	A ordem de 7 na lista acima (5)

## FUNÇÕES DE PROCURA E REFERÊNCIA

### COL

Retorna o número de coluna da referência especificada.

#### Sintaxe

**COL**(ref)

Ref é a célula ou intervalo de células cujo número da coluna você deseja obter.

- Se ref for omitido, presume-se que seja a referência da célula onde a função COL aparece.

Fórmula	Descrição (resultado)
=COL()	A coluna em que a fórmula aparece (1)
=COL(B10)	A coluna da referência (2)

### CORRESP

Retorna a posição relativa de um item em uma matriz que coincide com um valor especificado em uma ordem específica. Use CORRESP em vez de uma das funções PROC quando você precisar da posição de um item em um intervalo ao invés do item propriamente dito.

#### Sintaxe

**CORRESP**(valor\_procurado;matriz\_procurada;tipo\_correspondência)

Valor\_procurado é o valor utilizado para localizar o valor desejado em uma tabela.

- Valor\_procurado é o valor que você deseja coincidir com matriz\_procurada. Por exemplo, quando você procura o número do telefone de alguém em uma lista telefônica, você está usando o nome da pessoa como o valor da procura, mas o número do telefone é o valor que você deseja.
- Valor\_procurado pode ser um valor (número, texto ou valor lógico) ou uma referência de célula de um número, texto ou valor lógico.

Matriz\_procurada é um intervalo contíguo de células que contêm valores possíveis de procura. Matriz\_procurada precisa ser uma matriz ou uma referência de matriz.

Tipo\_correspondência é o número -1, 0 ou 1. Tipo\_correspondência especifica como o Microsoft Excel corresponde a valor\_procurado com os valores contidos em matriz\_procurada.

- Se tipo\_correspondência for 1, CORRESP localizará o maior valor que for menor do que ou igual a valor\_procurado. Matriz\_procurada deve ser posicionada em ordem ascendente: ...-2, -1, 0, 1, 2,...A-Z, FALSO, VERDADEIRO.
- Se tipo\_correspondência for 0, CORRESP localizará o primeiro valor que for exatamente igual a valor\_procurado. Matriz\_procurada pode ser colocada em qualquer ordem.
- Se tipo\_correspondência for -1, CORRESP localizará o menor valor que seja maior ou igual a valor\_procurado. Matriz\_procurada deve ser posicionada em ordem decrescente: VERDADEIRO, FALSO, Z-A,...2, 1, 0, -1, -2,... e assim por diante.
- Se tipo\_correspondência for omitido, será equivalente a 1.

#### Comentários

- CORRESP retorna a posição do valor coincidente em matriz\_procurada, e não o valor propriamente dito. Por exemplo: CORRESP("b";{"a"."b"."c"};0) retorna 2, a posição relativa de "b" na matriz {"a"."b"."c"}.
- CORRESP não faz distinção entre letras maiúsculas e minúsculas quando estiver fazendo a correspondência entre valores de texto.
- Se CORRESP não conseguir localizar um valor coincidente, ele fornecerá o valor de erro #N/D.
- Se tipo\_correspondência for 0 e valor\_procurado for um texto, valor\_procurado poderá conter os caracteres curinga asterisco (\*) e ponto de interrogação (?). Um asterisco corresponde a qualquer seqüência de caracteres; um ponto de interrogação corresponde a qualquer caractere individual.

	A	B
1	<b>Produto</b>	<b>Contagem</b>
2	Bananas	25
3	Laranjas	38
4	Maçãs	40
5	Pêras	41

<b>Fórmula</b>	<b>Descrição (resultado)</b>
=CORRESP(39;B2:B5;1)	Como não existe uma correspondência exata, é retornada a posição do próximo valor mais baixo (38) no intervalo B2:B5. (2)
=CORRESP(41;B2:B5;0)	A posição de 41 no intervalo B2:B5. (4)
=CORRESP(40;B2:B5;-1)	Retorna um erro porque o intervalo B2:B5 não está em ordem decrescente. #N/D

## ESCOLHER

Use ESCOLHER para selecionar um valor entre 29 valores que se baseie no número de índice. Por exemplo, se do valor1 até o valor7 forem os números da semana, ESCOLHER retorna um dos dias quando um número entre 1 e 7 for usado como núm\_índice. Use núm\_índice para retornar um valor da lista de argumentos de valor.

### Sintaxe

**ESCOLHER(núm\_índice;valor1;valor2,...)**

Núm\_índice especifica o argumento de valor selecionado. Núm\_índice deve ser um número entre 1 e 29, ou uma fórmula ou referência a uma célula que contenha um número entre 1 e 29.

- Se núm\_índice for 1, ESCOLHER retornará valor1; se for 2, ESCOLHER retornará valor2; e assim por diante.
- Se núm\_índice for menor do que 1 ou maior do que o número do último valor na lista, ESCOLHER retornará o valor de erro #VALOR!.

Valor1,valor2,... são 1 a 29 argumentos de valor a partir dos quais ESCOLHER seleciona um valor ou uma ação a ser realizada com base no núm\_índice. Os argumentos podem ser números, referências de célula, nomes definidos, fórmulas, funções ou texto.

Exemplo:

=ESCOLHER(2; "Maçã"; "Banana"; "Pêra") retorna Banana.

=ESCOLHER(DIA.DA.SEMANA("09/08/2006");"Dom";"Seg";"Ter";"Qua";"Qui";"Sex";"Sáb") retorna Qua, pois a função DIA.DA.SEMANA("09/08/2006") retorna 4 como resposta.

## INFODADOSTABELADINÂMICA

Retorna os dados armazenados em um relatório de tabela dinâmica. Você pode usar INFODADOSTABELADINÂMICA para recuperar dados de resumo de um relatório de tabela dinâmica, desde que esses dados de resumo estejam visíveis no relatório de tabela dinâmica.

**Observação** Você pode inserir rapidamente uma fórmula INFODADOSTABELADINÂMICA simples ao digitar = na célula em que você deseja que o valor seja retornado e ao clicar na célula do relatório da tabela dinâmica em que se encontram os dados que você deseja retornar.

### Sintaxe

**INFODADOSTABELADINÂMICA(campo\_de\_dados,tabela\_dinâmica,campo1,item1,campo2,item2,...)**

Campo\_de\_dados é o nome, entre aspas, do campo de dados em que se encontram os dados que você deseja recuperar.

Tabela\_dinâmica é uma referência a qualquer célula, intervalo de células ou intervalo nomeado de células de um relatório da tabela dinâmica. Essas informações são usadas para determinar o relatório da tabela dinâmica em que se encontram os dados que você deseja recuperar.

Campo1, Item1, Campo2, Item2 são um dos 14 pares de nomes de campo e item que descrevem os dados que você deseja recuperar. Os pares podem vir em qualquer ordem. Os nomes de campos e itens que não sejam datas e números vêm entre aspas.

Exemplo:

	A	B	C	D	E	F
1						
2						
3	Soma de Valor Vendido	Vendedor (a)				
4	Data da Venda	Ana Paula	Fernanda	Flávia	Paulo Henrique	Total geral
5	01/10/2004	172,8	154,8	138,5	132,9	599
6	02/10/2004	89,6	55,8	22,5	71,2	239,1
7	03/10/2004	119,1	95,3	152,1	105,59	472,09
8	04/10/2004	167,2	60,39	119,9	133	480,49
9	Total geral	548,7	366,29	433	442,69	1790,68

=**INFODADOSTABELADINÂMICA("Valor Vendido";\$A\$3)** retorna o total geral do campo Valor Total Vendido, ou seja, 1790,68.

=**INFODADOSTABELADINÂMICA("Valor Vendido";\$A\$3;"Data da Venda";DATA(2004;10;4))** retorna o total geral de Valor Vendido em 04/10/2004, ou seja, 480,49

=**INFODADOSTABELADINÂMICA("Valor Vendido";\$A\$3;"Data da Venda";DATA(2004;10;4);"Vendedor (a)";"Flávia")** retorna o total geral de Valor Vendido em 04/10/2004 da vendedora Flávia, ou seja, 119,90.

## LIN

Retorna o número da linha de uma referência.

### Sintaxe

#### LIN(ref)

Ref é a célula ou intervalo de células cujo número da linha você deseja obter.

- Se ref for omitido, será equivalente à referência da célula na qual a função LIN aparecer.

	A	B
1	<b>Fórmula</b>	<b>Descrição (Resultado)</b>
2	=LIN()	A linha em que a fórmula aparece (2)
3	=LIN(C10)	A linha da referência (10)

## PROCH

Localiza um valor específico na linha superior de uma tabela ou [matriz](#) de valores e retorna um valor na mesma coluna de uma linha especificada na tabela ou matriz. Use PROCH quando seus valores de comparação estiverem localizados em uma linha ao longo da parte superior de uma tabela de dados e você quiser observar um número específico de linhas mais abaixo. Use PROCV quando os valores de comparação estiverem em uma coluna à esquerda dos dados que você deseja localizar.

O H de PROCH significa "Horizontal."

### Sintaxe

#### PROCH(valor\_procurado;matriz\_tabela;núm\_índice\_lin;procurar\_intervalo)

Valor\_procurado é o valor a ser localizado na primeira linha da tabela. Valor\_procurado pode ser um valor, uma referência ou uma seqüência de caracteres de texto.

Matriz\_tabela é uma tabela de informações onde os dados devem ser procurados. Use uma referência para um intervalo ou um nome de intervalo.

- Os valores na primeira linha de matriz\_tabela podem ser texto, números ou valores lógicos.
- Se procurar\_intervalo for VERDADEIRO, os valores na primeira linha de matriz\_tabela deverão ser colocados em ordem ascendente: ...-2, -1, 0, 1, 2,... , A-Z, FALSO, VERDADEIRO, caso contrário, PROCH pode não retornar o valor correto. Se procurar\_intervalo for FALSO, matriz\_tabela não precisará ser ordenada.

- Textos em maiúsculas e minúsculas são equivalentes.

Núm\_índice\_lin é o número da linha em matriz\_tabela de onde o valor correspondente deve ser retirado. Um núm\_índice\_lin equivalente a 1 retorna o valor da primeira linha na matriz\_tabela, um núm\_índice\_lin equivalente a 2 retorna o valor da segunda linha na matriz\_tabela, e assim por diante. Se núm\_índice\_lin for menor do que 1, PROCH retornará o valor de erro #VALOR!; se núm\_índice\_lin for maior do que o número de linhas na matriz\_tabela, PROCH retornará o valor de erro #REF!.

Procurar\_intervalo é um valor lógico que especifica se você quer que PROCH localize uma correspondência exata ou aproximada. Se VERDADEIRO ou omitido, uma correspondência aproximada é retornada. Em outras palavras, se uma correspondência exata não for localizada, o valor maior mais próximo que seja menor que o valor\_procurado é retornada. Se FALSO, PROCH encontrará uma correspondência exata. Se nenhuma correspondência for localizada, o valor de erro #N/D será retornado.

#### Comentários

- Se PROCH não localizar valor\_procurado, e procurar\_intervalo for VERDADEIRO, ela usará o maior valor que é menor do que o valor\_procurado.
- Se o valor\_procurado for menor do que o menor valor na primeira linha de matriz\_tabela, PROCH retornará o valor de erro #N/D.

Fórmula	Descrição (resultado)
=PROCH("Eixos";A1:C4;2;VERDADEIRO)	Pesquisa Eixos na linha 1 e retorna o valor que está na linha 2 da mesma coluna (4)
=PROCH("Rolamentos";A1:C4;3;FALSO)	Pesquisa Rolamentos na linha 1 e retorna o valor que está na linha 3 da mesma coluna (7)
=PROCH("B";A1:C4;3;VERDADEIRO)	Pesquisa B na linha 1 e retorna o valor que está na linha 3 da mesma coluna. Como B não é uma coincidência exata, será usado o maior valor que seja inferior a B: Eixos (5)
=PROCH("Parafusos";A1:C4;4)	Pesquisa Parafusos na linha 1 e retorna o valor que está na linha 4 da mesma coluna (11)

	A	B	C
1	<b>Eixos</b>	<b>Rolamentos</b>	<b>Parafusos</b>
2	4	4	9
3	5	7	10
4	6	8	11

## PROCV

Localiza um valor na primeira coluna à esquerda de uma tabela e retorna um valor na mesma linha de uma coluna especificada na tabela. Use PROCV em vez de PROCH quando os valores da comparação estiverem posicionados em uma coluna à esquerda ou à direita dos dados que você deseja procurar.

O V em PROCV significa "Vertical".

#### Sintaxe

**PROCV(valor\_procurado;matriz\_tabela;núm\_índice\_coluna;procurar\_intervalo)**

Valor\_procurado é o valor a ser localizado na primeira coluna da [matriz](#). Valor\_procurado pode ser um valor, uma referência ou uma seqüência de caracteres de texto.

Matriz\_tabela é a tabela de informações em que os dados são procurados. Use uma referência para um intervalo ou nome de intervalo, tal como Banco de dados ou Lista.

- Se procurar\_intervalo for VERDADEIRO, os valores na primeira coluna de matriz\_tabela deverão ser colocados em ordem ascendente: ..., -2, -1, 0, 1, 2, ... , A-Z, FALSO, VERDADEIRO; caso contrário, PROCV pode não retornar o valor correto. Se procurar\_intervalo for FALSO, matriz\_tabela não precisará ser ordenada.
- Você pode colocar os valores em ordem ascendente escolhendo o comando **Classificar** no menu **Dados** e selecionando **Crescente**.
- Os valores na primeira coluna de matriz\_tabela podem ser texto, números ou valores lógicos.
- Textos em maiúsculas e minúsculas são equivalentes.

Núm\_índice\_coluna é o número da coluna em matriz\_tabela a partir do qual o valor correspondente deve ser retornado. Um núm\_índice\_coluna de 1 retornará o valor na primeira coluna em matriz\_tabela; um núm\_índice\_coluna de 2 retornará o valor na segunda coluna em matriz\_tabela, e assim por diante.

Se núm\_índice\_coluna for menor do que 1, PROCV retornará o valor de erro #VALOR!; se núm\_índice\_coluna for maior do que o número de colunas em matriz\_tabela, PROCV retornará o valor de erro #REF!.

Procurar\_intervalo é um valor lógico que especifica se você quer que PROCV encontre a correspondência exata ou uma correspondência aproximada. Se VERDADEIRO ou omitida, uma correspondência aproximada é retornada; em outras palavras, se uma correspondência exata não for encontrada, o valor maior mais próximo que é menor que o valor\_procurado é retornado. Se FALSO, PROCV encontrará uma correspondência exata. Se nenhuma correspondência for encontrada, o valor de erro #N/D é retornado.

### Comentários

- Se PROCV não localizar valor\_procurado e procurar\_intervalo for VERDADEIRO, ela usará o maior valor que for menor do que ou igual a valor\_procurado.
- Se valor\_procurado for menor do que o menor valor na primeira coluna de matriz\_tabela, PROCV fornecerá o valor de erro #N/D.
- Se PROCV não localizar valor\_procurado e procurar\_intervalo for FALSO, PROCV fornecerá o valor #N/D.

Fórmula	Descrição (resultado)
=PROCV(1;A2:C10;2)	Pesquisa 1 na coluna A e retorna o valor que está na mesma linha na coluna B (2,17)
=PROCV(1;A2:C10;3;VERDADEIRO)	Pesquisa 1 na coluna A e retorna o valor que está na mesma linha na coluna C (100)
=PROCV(0,7;A2:C10;3;FALSO)	Pesquisa 0,746 na coluna A. Como não existe correspondência exata na coluna A, é retornado um erro (#N/D)
=PROCV(0,1;A2:C10;2;VERDADEIRO)	Pesquisa 0,1 na coluna A. Como 0,1 é inferior ao menor valor da coluna A, é retornado um erro (#N/D)
=PROCV(2;A2:C10;2;VERDADEIRO)	Pesquisa 2 na coluna A e retorna o valor que está na mesma linha na coluna B (1,71)

	A	B	C
1	Densidade	Viscosidade	Temperatura
2	0,457	3,55	500
3	0,525	3,25	400
4	0,616	2,93	300
5	0,675	2,75	250
6	0,746	2,57	200
7	0,835	2,38	150
8	0,946	2,17	100
9	1,09	1,95	50
10	1,29	1,71	0

## TRANSPOR

Retorna um intervalo vertical de células como um intervalo horizontal ou vice-versa. A função TRANSPOR precisa ser inserida como uma fórmula de matriz em um intervalo com o mesmo número de linhas e colunas, respectivamente, das existentes em uma matriz. Use TRANSPOR para deslocar a orientação vertical e horizontal de uma matriz em uma planilha.

### Sintaxe

#### TRANSPOR(matriz)

Matriz é uma matriz ou um intervalo de células em uma planilha que se deseja transpor. A transposição de uma matriz é criada utilizando-se a primeira linha da matriz como a primeira coluna da nova matriz, a segunda linha da matriz como a segunda coluna da nova matriz, e assim por diante.

Exemplo:

	A	B	C
1	Produto	Preço	Quantidade
2	DVD	R\$ 199,00	5
3	TV	R\$ 399,00	2
4	Impressora	R\$ 180,00	3

=TRANSPOR(A2:C2)

---

**Observação:** A fórmula no exemplo precisa ser inserida como uma fórmula de matriz. Após copiar o exemplo em uma planilha em branco, digite a fórmula na célula A7. Depois selecione o intervalo A7:A9, pressione F2 e, em seguida, pressione CTRL+SHIFT+ENTER.

Se quiser transpor a planilha inteira digite, por exemplo, na célula A11, a fórmula =TRANSPOR(A1:C4). Depois selecione as células A11:D13, pressione F2 e, em seguida, pressione CTRL+SHIFT+ENTER.

## FUNÇÕES DE BANCO DE DADOS

### **BDCONTAR**

Conta as células contendo números em uma coluna de uma lista ou de um banco de dados que corresponde às condições especificadas por você.

O argumento de campo é opcional. Se o campo for omitido, BDCONTAR contará todos os registros no banco de dados que coincidirem com os critérios.

#### **Sintaxe**

**BDCONTAR(banco\_dados;campo;critérios)**

**Banco\_dados** é o intervalo de células que constitui a lista ou o banco de dados. Um banco de dados é uma lista de dados relacionados na qual as linhas de informações relacionadas são os registros e as colunas de dados são os campos. A primeira linha da lista contém os rótulos de cada coluna.

**Campo** indica a coluna que será usada na função. O campo pode ser dado como texto com o rótulo da coluna entre aspas, como "Idade" ou "Rendimento", ou como um número que represente a posição da coluna dentro da lista: 1 para a primeira coluna, 2 para a segunda coluna e assim por diante.

**Critérios** é o intervalo de células que contém as condições especificadas. Você pode usar qualquer intervalo para o argumento de critérios, desde que ele inclua pelo menos um rótulo de coluna e ao menos uma célula abaixo do rótulo de coluna para especificar uma condição para a coluna.

### **BDCONTARA**

Conta as células que não estão em branco em uma coluna de uma lista ou de um banco de dados que corresponde às condições especificadas por você.

O argumento de campo é opcional. Se o campo for omitido, BDCONTARA contará todos os registros no banco de dados que coincidirem com os critérios.

#### **Sintaxe**

**BDCONTARA(banco\_dados;campo;critérios)**

**Banco\_dados** é o intervalo de células que constitui a lista ou o banco de dados. Um banco de dados é uma lista de dados relacionados na qual as linhas de informações relacionadas são os registros e as colunas de dados são os campos. A primeira linha da lista contém os rótulos de cada coluna.

**Campo** indica a coluna que será usada na função. O campo pode ser dado como texto com o rótulo da coluna entre aspas, como "Idade" ou "Rendimento", ou como um número que represente a posição da coluna dentro da lista: 1 para a primeira coluna, 2 para a segunda coluna e assim por diante.

**Critérios** é o intervalo de células que contém as condições especificadas. Você pode usar qualquer intervalo para o argumento de critérios, desde que ele inclua pelo menos um rótulo de coluna e ao menos uma célula abaixo do rótulo de coluna para especificar uma condição para a coluna.

### **BDEXTRAIR**

Extrai um único valor de uma coluna em uma lista ou banco de dados que corresponde às condições especificadas por você.

#### **Sintaxe**

**BDEXTRAIR (banco\_dados;campo;critérios)**

---

**Banco\_dados** é o intervalo de células que constitui a lista ou o banco de dados. Um banco de dados é uma lista de dados relacionados na qual as linhas de informações relacionadas são os registros e as colunas de dados são os campos. A primeira linha da lista contém os rótulos de cada coluna.

**Campo** indica a coluna que será usada na função. O campo pode ser dado como texto com o rótulo da coluna entre aspas, como "Idade" ou "Rendimento", ou como um número que represente a posição da coluna dentro da lista: 1 para a primeira coluna, 2 para a segunda coluna e assim por diante.

**Critérios** é o intervalo de células que contém as condições especificadas. Você pode usar qualquer intervalo para o argumento de critérios, desde que ele inclua pelo menos um rótulo de coluna e ao menos uma célula abaixo do rótulo de coluna para especificar uma condição para a coluna.

#### **Comentários**

- Se nenhum registro coincidir com os critérios, BDEXTRAIR retornará o valor de erro #VALOR!.
- Se mais de um registro coincidir com os critérios, BDEXTRAIR retornará o valor de erro #NÚM!.

### **BDMÁX**

Retorna o maior número de uma coluna em uma lista ou banco de dados que coincide com as condições especificadas por você.

#### **Sintaxe**

##### **BDMÁX(banco\_dados;campo;critérios)**

**Banco\_dados** é o intervalo de células que constitui a lista ou o banco de dados. Um banco de dados é uma lista de dados relacionados na qual as linhas de informações relacionadas são os registros e as colunas de dados são os campos. A primeira linha da lista contém os rótulos de cada coluna.

**Campo** indica a coluna que será usada na função. O campo pode ser dado como texto com o rótulo da coluna entre aspas, como "Idade" ou "Rendimento", ou como um número que represente a posição da coluna dentro da lista: 1 para a primeira coluna, 2 para a segunda coluna e assim por diante.

**Critérios** é o intervalo de células que contém as condições especificadas. Você pode usar qualquer intervalo para o argumento de critérios, desde que ele inclua pelo menos um rótulo de coluna e ao menos uma célula abaixo do rótulo de coluna para especificar uma condição para a coluna.

### **BDMÉDIA**

Calcula a média dos valores em uma coluna de uma lista ou banco de dados que corresponde às condições especificadas por você.

#### **Sintaxe**

##### **BDMÉDIA(banco\_dados;campo;critérios)**

**Banco\_dados** é o intervalo de células que constitui a lista ou o banco de dados. Um banco de dados é uma lista de dados relacionados na qual as linhas de informações relacionadas são os registros e as colunas de dados são os campos. A primeira linha da lista contém os rótulos de cada coluna.

**Campo** indica a coluna que será usada na função. O campo pode ser dado como texto com o rótulo da coluna entre aspas, como "Idade" ou "Rendimento", ou como um número que represente a posição da coluna dentro da lista: 1 para a primeira coluna, 2 para a segunda coluna e assim por diante.

**Critérios** é o intervalo de células que contém as condições especificadas. Você pode usar qualquer intervalo para o argumento de critérios, desde que ele inclua pelo menos um rótulo de coluna e ao menos uma célula abaixo do rótulo de coluna para especificar uma condição para a coluna.

### **BDMÍN**

Retorna o menor número de uma coluna em uma lista ou banco de dados que coincide com as condições especificadas por você.

#### **Sintaxe**

##### **BDMÍN(banco\_dados;campo;critérios)**

**Banco\_dados** é o intervalo de células que constitui a lista ou o banco de dados. Um banco de dados é uma lista de dados relacionados na qual as linhas de informações relacionadas são os registros e as colunas de dados são os campos. A primeira linha da lista contém os rótulos de cada coluna.

**Campo** indica a coluna que será usada na função. O campo pode ser dado como texto com o rótulo da coluna entre aspas, como "Idade" ou "Rendimento", ou como um número que represente a posição da coluna dentro da lista: 1 para a primeira coluna, 2 para a segunda coluna e assim por diante.

**Crítérios** é o intervalo de células que contém as condições especificadas. Você pode usar qualquer intervalo para o argumento de critérios, desde que ele inclua pelo menos um rótulo de coluna e ao menos uma célula abaixo do rótulo de coluna para especificar uma condição para a coluna.

### **BDMULTIPL**

Multiplica os valores de uma coluna em uma lista ou banco de dados que coincidem com as condições especificadas por você.

#### **Sintaxe**

**BDMULTIPL(banco\_dados;campo;critérios)**

**Banco\_dados** é o intervalo de células que constitui a lista ou o banco de dados. Um banco de dados é uma lista de dados relacionados na qual as linhas de informações relacionadas são os registros e as colunas de dados são os campos. A primeira linha da lista contém os rótulos de cada coluna.

**Campo** indica a coluna que será usada na função. O campo pode ser dado como texto com o rótulo da coluna entre aspas, como "Idade" ou "Rendimento", ou como um número que represente a posição da coluna dentro da lista: 1 para a primeira coluna, 2 para a segunda coluna e assim por diante.

**Crítérios** é o intervalo de células que contém as condições especificadas. Você pode usar qualquer intervalo para o argumento de critérios, desde que ele inclua pelo menos um rótulo de coluna e ao menos uma célula abaixo do rótulo de coluna para especificar uma condição para a coluna.

### **BDSOMA**

Soma os números de uma coluna em uma lista ou banco de dados que coincidem com as condições especificadas por você.

#### **Sintaxe**

**BDSOMA(banco\_dados;campo;critérios)**

**Banco\_dados** é o intervalo de células que constitui a lista ou o banco de dados. Um banco de dados é uma lista de dados relacionados na qual as linhas de informações relacionadas são os registros e as colunas de dados são os campos. A primeira linha da lista contém os rótulos de cada coluna.

**Campo** indica a coluna que será usada na função. O campo pode ser dado como texto com o rótulo da coluna entre aspas, como "Idade" ou "Rendimento", ou como um número que represente a posição da coluna dentro da lista: 1 para a primeira coluna, 2 para a segunda coluna e assim por diante.

**Crítérios** é o intervalo de células que contém as condições especificadas. Você pode usar qualquer intervalo para o argumento de critérios, desde que ele inclua pelo menos um rótulo de coluna e ao menos uma célula abaixo do rótulo de coluna para especificar uma condição para a coluna.

#### **Exemplos:**

	A	B	C	D	E	F
1	<b>Árvore</b>	<b>Altura</b>	<b>Idade</b>	<b>Produção</b>	<b>Lucro</b>	<b>Altura</b>
2	Maçã	>10				<16
3	Pêra					
4	<b>Árvore</b>	<b>Altura</b>	<b>Idade</b>	<b>Produção</b>	<b>Lucro</b>	
5	Maçã	18	20	14	105	
6	Pêra	12	12	10	96	
7	Cereja	13	14	9	105	
8	Maçã	14	15	10	75	
9	Pêra	9	8	8	76,8	
10	Maçã	8	9	6	45	

<b>Fórmula</b>	<b>Descrição (resultado)</b>
=BDCONTAR(A4:E10;"Idade";A1:F2)	Esta função analisa os registros de macieiras com altura entre 10 e 16 e conta quantos campos Idade nesses registros contêm números (1)
=BDCONTARA(A4:E10;"Lucro";A1:F2)	Esta função analisa os registros de macieiras com altura entre 10 e

	16 e conta quantos campos Lucro nesses registros não estão em branco (1)
=BDMÁX(A4:E10;"Lucro";A1:A3)	O lucro máximo das macieiras e pereiras (105)
=BDMÍN(A4:E10;"Lucro";A1:B2)	O lucro mínimo das macieiras com altura acima de 10 (75)
=BDSOMA(A4:E10;"Lucro";A1:A2)	O lucro total das macieiras (225)
=BDSOMA(A4:E10;"Lucro";A1:F2)	O lucro total das macieiras com altura entre 10 e 16 (75)
=BDMULTIPL(A4:E10;"Produção";A1:B2)	O produto das produções das macieiras com altura acima de 10 (140)
=BDMÉDIA(A4:E10;"Produção";A1:B2)	A produção média das macieiras com altura acima de 10 pés (12)
=BDMÉDIA(A4:E10;3;A4:E10)	A idade média de todas as árvores no banco de dados (13)
=BDEXTRAI(A4:E10;"Produção";A1:A3)	Retornará o valor de erro #NÚM! porque mais de um registro coincide com os critérios

## DiCAS

- Você pode usar qualquer intervalo para o argumento de critérios, desde que ele inclua pelo menos um rótulo de coluna e ao menos uma célula abaixo do rótulo de coluna para especificar a condição. Por exemplo, se o intervalo G1:G2 contiver o rótulo de coluna Receita em G1 e a quantia 10.000 em G2, você poderá definir o intervalo como CoincidirReceita e usar este nome como o argumento de critérios nas funções de banco de dados.
- Embora o intervalo de critérios possa ser colocado em qualquer local na planilha, não o posicione abaixo da lista. Se você adicionar outras informações à lista, usando o comando **Formulário** do menu **Dados**, as novas informações serão adicionadas na primeira linha abaixo da lista. Se a linha abaixo da lista não estiver em branco, o Microsoft Excel não poderá adicionar as novas informações.
- Certifique-se de que o intervalo de critérios não se sobreponha à lista.
- Para efetuar uma operação em uma coluna inteira em um banco de dados, insira uma linha em branco abaixo dos rótulos de coluna no intervalo de critérios.

## Exemplos de critérios

### Várias condições em uma só coluna

<b>Vendedor</b>
Ribeiro
Cardoso
Rodrigo

Se você tiver duas ou mais condições para uma única coluna, digite os critérios um diretamente abaixo do outro em linhas separadas. Por exemplo, o intervalo de critérios ao lado exibe as linhas que contêm "Ribeiro", "Cardoso" ou "Rodrigo" na coluna Vendedor.

### Uma condição em duas ou mais colunas

<b>Tipo</b>	<b>Vendedor</b>	<b>Vendas</b>
Produção	Ribeiro	>1000

Para localizar dados que atendam a uma condição em duas ou mais colunas, insira todos os critérios na mesma linha do intervalo de critérios. Por exemplo, o intervalo de critérios ao lado exibe todas as linhas que contêm "Produção" na coluna Tipo, "Ribeiro" na coluna Vendedor e valores de vendas acima de R\$ 1.000.

### Uma condição em uma coluna ou outra

<b>Tipo</b>	<b>Vendedor</b>	<b>Vendas</b>
Produção		
	Ribeiro	
		>1000

Para localizar os dados que atendam a uma condição em uma coluna ou uma condição em outra coluna, insira os critérios em linhas diferentes do intervalo de critérios. Por exemplo, o intervalo de critérios ao lado exibe todas as linhas que contêm "Produção" na coluna Tipo, "Ribeiro" na coluna Vendedor ou valores de vendas acima de R\$ 1.000.

### Um de dois conjuntos de condições para duas colunas

Vendedor	Vendas
Ribeiro	>3000
Cardoso	>1500

Para localizar linhas que atendam a um de dois conjuntos de condições, onde cada conjunto inclui condições para mais de uma coluna, digite os critérios em linhas separadas. Por exemplo, o intervalo de critérios ao lado exibe as linhas que contêm tanto "Ribeiro" na coluna Vendedor e valores de vendas acima de R\$ 3.000, como também exibe as linhas do vendedor Cardoso com valores de vendas acima de R\$ 1.500.

### **Mais de dois conjuntos de condições para uma coluna**

Vendas	Vendas
>5000	<8000
<500	

Para localizar as linhas que atendem a mais de dois conjuntos de critérios, inclua várias colunas com o mesmo título de coluna. Por exemplo, o intervalo de critérios ao lado exibe as vendas entre R\$ 5.000 e R\$ 8.000 além de vendas inferiores a R\$ 500.

### **Condições criadas como resultado de uma fórmula**

=C7>MÉDIA(\$C\$7:\$C\$10)
---------------------------

Você pode usar um valor calculado que seja o resultado de uma fórmula como critério. Ao usar uma fórmula para criar um critério, não use um rótulo de coluna como rótulo de critérios; deixe o rótulo de critérios em branco ou use um rótulo que não seja um rótulo de coluna na lista. Por exemplo, o intervalo de critérios ao lado exibe linhas com um valor na coluna C maior que a média das células C7:C10.

### **Observações**

- A fórmula utilizada para uma condição deve usar uma referência relativa para se referir ao rótulo de coluna (por exemplo, Vendas) ou o campo correspondente no primeiro registro. Todas as outras referências na fórmula devem ser referências absolutas e a fórmula deve ser avaliada como VERDADEIRO ou FALSO. Na fórmula de exemplo, "C7" se refere ao campo (coluna C) do primeiro registro (linha 7) da lista.
- Você pode usar um rótulo de coluna na fórmula, em vez de uma referência relativa de célula ou um nome de intervalo. Quando o Microsoft Excel exibe um valor de erro como #NOME? ou #VALOR! na célula que contém o critério, você pode ignorar este erro porque ele não afeta a maneira como a lista é filtrada.
- Ao avaliar dados, o Microsoft Excel não distingue entre caracteres maiúsculos ou minúsculos.

## **FUNÇÕES DE TEXTO**

### **ARRUMAR**

Remove todos os espaços do texto exceto os espaços únicos entre palavras. Use ARRUMAR no texto que recebeu de outro aplicativo que pode ter espaçamento irregular.

#### **Sintaxe**

#### **ARRUMAR(texto)**

Texto é o texto do qual se deseja remover espaços.

Exemplo:

=ARRUMAR(" **Ganhos do primeiro trimestre** ") remove os espaços precedentes e posteriores do texto na fórmula (Ganhos do primeiro trimestre)

### **CONCATENAR**

Agrupa várias seqüências de caracteres de texto em uma única seqüência de caracteres de texto.

#### **Sintaxe**

#### **CONCATENAR (texto1;texto2; ...)**

Texto1; texto2;... são 1 a 30 itens de texto a serem agrupados em um único item de texto. Os itens de texto podem ser seqüência de caracteres de texto, números ou referências de célula única.

#### **Comentários**

O operador "&" pode ser usado no lugar de CONCATENAR para agrupar itens de texto.

	A	Exemplo:
1	Dados	=CONCATENAR("A população do rio de ";A3;" ";A2;" é de ";A4;" /m") concatena uma frase a partir dos dados acima (A população do rio de espécie truta é de 32/m)
2	truta	
3	espécie	
4	32	

## DIREITA

DIREITA retorna o último caractere ou caracteres em uma seqüência de caracteres de texto com base no número de caracteres especificado por você.

### Sintaxe

**DIREITA(texto;núm\_caract)**

Texto é a seqüência de caracteres de texto que contém os caracteres que você deseja extrair.

Núm\_caract especifica o número de caracteres a ser extraído por DIREITA.

### Comentários

- Núm\_caract deve ser maior ou igual a zero.
- Se núm\_caract for maior do que o comprimento do texto, DIREITA retornará todo o texto.
- Se núm\_caract for omitido, será considerado 1.

Exemplo:

	A	B	
1	People Computação		=DIREITA(A1;4) resulta nos 4 últimos caracteres de "People Computação", ou seja, "ação".
2	TEQI-1830-2006/01		=DIREITA(A1) resulta "o".

## ESQUERDA

ESQUERDA retorna o primeiro caractere ou caracteres em uma seqüência de caracteres de texto baseado no número de caracteres especificado por você.

### Sintaxe

**ESQUERDA(texto;núm\_caract)**

Texto é a seqüência de caracteres de texto que contém os caracteres que você deseja extrair.

Núm\_caract especifica o número de caracteres que você deseja que ESQUERDA extraia.

- Núm\_caract deve ser maior ou igual a zero.
- Se núm\_caract for maior do que o comprimento do texto, ESQUERDA retornará todo o texto.
- Se núm\_caract for omitido, será considerado 1.

	A	B	Exemplo:
1	People Computação		=ESQUERDA(A1;6) resulta nos 6 primeiros caracteres de "People Computação", ou seja, "People"
2	TEQI-1830-2006/01		

## EXATO

Compara duas seqüências de caracteres de texto e retorna VERDADEIRO se elas forem exatamente iguais e FALSO caso contrário. EXATO faz diferenciação entre maiúsculas e minúsculas, mas ignora diferenças de formatação. Use EXATO para testar o texto inserido em um documento.

### Sintaxe

**EXATO(texto1,texto2)**

Texto1 é a primeira seqüência de caracteres de texto.

Texto2 é a segunda seqüência de caracteres de texto.

Exemplo:

	A	B	Fórmula	Descrição (resultado)
1	Primeira seqüência de caracteres	Segunda seqüência de caracteres		
2	palavra	palavra	=EXATO(A2;B2)	Verifica se as seqüências de caracteres da primeira linha coincidem (VERDADEIRO)
3	Palavra	palavra	=EXATO(A3;B3)	Verifica se as seqüências de caracteres da segunda linha coincidem (FALSO)
4	p palavra	palavra	=EXATO(A4;B4)	Verifica se as seqüências de caracteres da terceira linha coincidem (FALSO)

## EXT.TEXTO

EXT.TEXTO retorna um número específico de caracteres da seqüência de caracteres texto, começando na posição especificada, com base no número de caracteres especificado.

### Sintaxe

**EXT.TEXTO(texto;núm\_inicial;núm\_caract)**

Texto é a seqüência de caracteres de texto que contém os caracteres que você deseja extrair.

Núm\_inicial é a posição do primeiro caractere que você deseja extrair como texto. O primeiro caractere em texto possui núm\_inicial1, e assim por diante.

Núm\_caract especifica o número de caracteres que você deseja que EXT.TEXTO retorne do texto.

### Comentários

- Se núm\_inicial for maior do que o comprimento de texto, EXT.TEXTO retornará "" (texto vazio).
- Se núm\_inicial for menor do que o comprimento de texto, mas núm\_inicial e núm\_caract excederem o comprimento de texto, EXT.TEXTO retornará os caracteres até o final do texto.
- Se núm\_inicial for menor do que 1, EXT.TEXTO retornará o valor de erro #VALOR!.
- Se núm\_caract for negativo, EXT.TEXTO retornará o valor de erro #VALOR!.

	A
1	<b>Dados</b>
2	Fluxo de fluido

Fórmula	Descrição (resultado)
=EXT.TEXTO(A2;1;5)	Cinco caracteres da seqüência de caracteres acima, iniciando no primeiro caractere (Fluxo)
=EXT.TEXTO(A2;7;20)	Vinte caracteres da seqüência de caracteres acima, iniciando no sétimo (de fluido)
=EXT.TEXTO(A2;20;5)	Como o ponto inicial é maior do que o comprimento da seqüência de caracteres, será retornado texto vazio ()

## MAIÚSCULA

Converte o texto em maiúsculas.

### Sintaxe

**MAIÚSCULA(texto)**

Texto é o texto que se deseja converter para maiúsculas. Texto pode ser uma referência ou uma seqüência de caracteres de texto.

	A
1	<b>Dados</b>
2	total
3	Rendimento

Fórmula	Descrição (resultado)
=MAIÚSCULA(A2)	Coloca em maiúsculas a primeira seqüência de caracteres (TOTAL)
=MAIÚSCULA(A3)	Coloca em maiúsculas a segunda seqüência de caracteres (RENDIMENTO)

## MINÚSCULA

Converte todas as letras maiúsculas em uma seqüência de caracteres de texto para minúsculas.

### Sintaxe

**MINÚSCULA(texto)**

Texto é o texto que você deseja converter para minúscula. MINÚSCULA só muda caracteres de letras para texto.

	A
1	People Computação

=MINÚSCULA(A1) coloca em minúsculas o texto "People Computação", ou seja, "people computação"

## MOEDA

A função descrita neste tópico da Ajuda converte um número em formato de texto e aplica um símbolo de unidade monetária. O nome da função (e o símbolo aplicado) depende das suas configurações de idioma.

Converte núm em texto usando o formato moeda, com decimais arredondado para o valor especificado. O formato usado é \$#.##0,00\_);(\$#.##0,00).

### Sintaxe

## MOEDA(núm;decimais)

Núm é um número, uma referência a uma célula contendo um número ou uma fórmula que avalia um número.

Decimais é o número de dígitos à direita da vírgula decimal. Se decimais for negativo, núm será arredondado à esquerda da vírgula decimal. Se omitir decimais, será considerado 2.

### Comentários

A principal diferença entre formatar uma célula contendo um número com o comando **Células** (menu **Formatar**) e formatar um número diretamente com a função MOEDA é que ela converte o resultado em texto. Um número formatado com o comando **Células** continua a ser um número. Você pode continuar a usar números formatados com MOEDA nas fórmulas, porque o Microsoft Excel converte os números inseridos como valores de texto em números quando calcula.

A		Fórmula	Descrição (resultado)
1	Dados	=MOEDA(A2; 2)	Exibe o primeiro número em um formato de moeda, 2 dígitos à direita da vírgula decimal (R\$ 1.234,57)
2	1234,567		
3	-1234,57	=MOEDA(A2; -2)	Exibe o primeiro número em um formato de moeda, 2 dígitos à esquerda da vírgula decimal (R\$ 1.200)
4	-0,123	=MOEDA(A3; -2)	Exibe o segundo número em um formato de moeda, 2 dígitos à esquerda da vírgula decimal ((R\$ 1.200))
5	99,888	=MOEDA(A4; 4)	Exibe o terceiro número em um formato de moeda, 4 dígitos à direita da vírgula decimal ((R\$ 0,1230))
		=MOEDA(A5)	Exibe o quarto número em um formato de moeda, 2 dígitos à esquerda da vírgula decimal (R\$ 99,89)

## MUDAR

MUDAR substitui parte de uma seqüência de caracteres de texto, com base no número de caracteres especificado, por uma seqüência diferente.

### Sintaxe

**MUDAR(texto\_antigo;núm\_inicial;núm\_caract;novo\_texto)**

A		
1	Dados	Texto_antigo é o texto no qual você deseja trocar alguns caracteres.
2	abcdefghijkl	Núm_inicial é a posição do caractere em texto_antigo que você deseja substituir por novo_texto.
3	2009	Núm_caract é o número de caracteres em texto_antigo que você deseja que MUDAR substitua por novo_texto.
4	123456	Novo_texto é o texto que substituirá os caracteres em texto_antigo.

Fórmula	Descrição (resultado)
=MUDAR(A2;6;5;"*")	Substitui cinco caracteres, iniciando no sexto caractere (abcde*k)
=MUDAR(A3;3;2;"10")	Substitui os dois últimos dígitos de 2009 por 10 (2010)
=MUDAR(A4;1;3;"@")	Substitui os três primeiros caracteres por @ (@456)

## NÚM.CARACT

NÚM.CARACT retorna o número de caracteres em uma seqüência de caracteres de texto.

### Sintaxe

**NÚM.CARACT(texto)**

Texto é o texto cujo tamanho se deseja determinar. Os espaços contam como caracteres.

A		B	
1	People	Fórmula	Descrição (resultado)
2	Computação	=NÚM.CARACT(A1)	O comprimento da primeira seqüência de caracteres (17)
	Informática	=NÚM.CARACT(A3)	O comprimento da segunda seqüência de caracteres (11)

## PROCURAR

Coloca a primeira letra de uma seqüência de caracteres de texto em maiúscula e todas as outras letras do texto depois de qualquer caractere diferente de uma letra. Converte todas as outras letras para minúsculas.

### Sintaxe

#### PROCURAR(texto)

Texto é o texto entre aspas, uma fórmula que retorna o texto ou uma referência a uma célula que contenha o texto que você deseja colocar parcialmente em maiúscula.

	A
1	<b>Dados</b>
2	isto é um TÍTULO
3	vale 2 centavos
4	76Orçamento

Fórmula	Descrição (resultado)
=PROCURAR("I";A2)	Primeira letra maiúscula na primeira seqüência de caracteres (Isto É Um Título)
=PROCURAR("V";A3)	Primeira letra maiúscula na segunda seqüência de caracteres (Vale 2 Centavos)
=PROCURAR("O";A4)	Primeira letra maiúscula na terceira seqüência de caracteres (76Orçamento)

## PROCURAR

PROCURAR localiza uma seqüência de caracteres de texto (texto\_procurado) em outra seqüência (no\_texto) e retorna o número da posição inicial de texto\_procurado a partir do primeiro caractere de no\_texto. Você também pode usar LOCALIZAR para encontrar uma seqüência de caracteres de texto em outra, mas ao contrário de LOCALIZAR, PROCURAR diferencia maiúsculas e minúsculas e não permite caracteres curinga.

### Sintaxe

#### PROCURAR(texto\_procurado;no\_texto;núm\_inicial)

Texto\_procurado é o texto que se deseja localizar.

No\_texto é o texto que contém o texto que se deseja localizar.

Núm\_inicial especifica o caractere a partir do qual a pesquisa será iniciada. O primeiro caractere no\_texto é o caractere número 1. Se você omitir núm\_inicial, será considerado como 1.

### Comentários

- Se texto\_procurado for "" (texto vazio), PROCURAR coincide com o primeiro caractere na pesquisa de seqüência de caracteres (ou seja, o caractere numerado núm\_inicial ou 1).
- Texto\_procurado não pode conter qualquer caractere curinga.
- Se texto\_procurado não aparecer no\_texto, PROCURAR retornará o valor de erro #VALOR!.
- Se núm\_inicial não for maior do que zero, PROCURAR retornará o valor de erro #VALOR!.
- Se núm\_inicial for maior do que o comprimento de no\_texto, PROCURAR retornará o valor de erro #VALOR!.

	A
1	<b>Dados</b>
2	Miriam Machado

Fórmula	Descrição (resultado)
=PROCURAR("M";A2)	A posição do primeiro "M" na seqüência de caracteres acima (1)
=PROCURAR("m";A2)	A posição do primeiro "m" na seqüência de caracteres acima (6)
=PROCURAR("M";A2,3)	A posição do primeiro "M" na seqüência de caracteres acima, iniciando com o terceiro caractere (8)

## REPT

Repete um texto um determinado número de vezes. Utilize REPT para preencher uma célula com um número de ocorrências de uma seqüência de caracteres de texto.

## Sintaxe

### REPT(texto;núm\_vezes)

Texto é o texto que você deseja repetir.

Núm\_vezes é um número positivo que especifica o número de vezes que você deseja repetir texto.

### Comentários

- Se núm\_vezes for 0 (zero), REPT retornará "" (texto vazio).
- Se núm\_vezes não for um inteiro, será truncado.
- O resultado da função REPT não pode ter mais de 32.767 caracteres ou REPT retornará #VALOR!.

Fórmula	Descrição (resultado)
=REPT("*-", 3)	Exibe a seqüência de caracteres 3 vezes (*-*-*-)
=REPT("-",10)	Exibe um traço 10 vezes (-----)

## SUBSTITUIR

Coloca novo\_texto no lugar de texto\_antigo em uma seqüência de caracteres de texto. Use SUBSTITUIR quando quiser substituir texto específico em uma seqüência de caracteres de texto; use MUDAR quando quiser substituir qualquer texto que ocorra em um local específico de uma seqüência de caracteres de texto.

### Sintaxe

#### SUBSTITUIR(texto;texto\_antigo;novo\_texto;núm\_da\_ocorrência)

Texto é o texto ou a referência a uma célula que contém o texto no qual deseja substituir caracteres.

Texto\_antigo é o texto que se deseja substituir.

Novo\_texto é o texto pelo qual deseja substituir texto\_antigo.

Núm\_da\_ocorrência especifica que ocorrência de texto\_antigo se deseja substituir por novo\_texto. Se especificar núm\_da\_ocorrência, apenas aquela ocorrência de texto\_antigo será substituída. Caso contrário, cada ocorrência de texto\_antigo em texto é alterada para novo\_texto.

	A
1	<b>Dados</b>
2	Dados de Vendas
3	Trimestre 1, 2008
4	Trimestre 1, 2011

Fórmula	Descrição (resultado)
=SUBSTITUIR(A2; "Vendas"; "Custo")	Substitui Vendas por Custo (Dados de Custo)
=SUBSTITUIR(A3; "1"; "2"; 1)	Substitui a primeira ocorrência de "1" por "2" (Trimestre 2, 2008)
=SUBSTITUIR(A4; "1"; "2"; 3)	Substitui a terceira ocorrência de "1" por "2" (Trimestre 1, 2012)

## TEXTO

Converte um valor para texto em um formato de número específico.

### Sintaxe

#### TEXTO(valor;format\_texto)

Valor é um valor numérico, uma fórmula que avalia para um valor numérico, ou uma referência a uma célula que contém um valor numérico.

Format\_texto é um formato de número na forma de texto contido na caixa **Categoria** da guia **Número** na caixa de diálogo **Formatar células**.

### Comentários

- Format\_texto não pode conter um asterisco (\*).
- Formatar a célula com uma opção na guia **Número** (comando **Células**, menu **Formatar**) altera apenas o formato, não o valor. Usar a função TEXTO converte um valor para texto formatado, e o resultado não é mais calculado como um número.
-

	A	B
1	<b>Vendedor</b>	<b>Vendas</b>
2	Cardoso	2800
3	Couto	40%

Fórmula	Descrição (resultado)
=A2&" vendeu o equivalente a "&TEXTO(B2; "R\$ 0,00")&" em unidades."	Combina o conteúdo acima em uma frase (Cardoso vendeu o equivalente a R\$ 2.800,00 em unidades.)
=A3&" vendeu "&TEXTO(B3;"0%")&" da venda total."	Combina o conteúdo acima em uma frase (Couto vendeu 40% da venda total.)

## FUNÇÕES LÓGICAS

### E

Retornará VERDADEIRO se todos os argumentos forem verdadeiros; retornará FALSO se um ou mais argumentos forem falsos. Essa função é muito usada juntamente com a função SE.

#### Sintaxe

E(lógico1;lógico2; ...)

Lógico1; lógico2;... são de 1 a 30 condições que você deseja testar e que podem ser VERDADEIRO ou FALSO.

#### Comentários

- Os argumentos devem ser avaliados para valores lógicos, como VERDADEIRO ou FALSO, ou devem ser [matrizes](#) ou referências que contêm valores lógicos.
- Se um argumento de uma matriz ou referência contiver texto ou células vazias, esses valores serão ignorados.
- Se o intervalo especificado não contiver valores lógicos, E retornará o valor de erro #VALOR!.

Fórmula	Descrição (resultado)
=E(VERDADEIRO; VERDADEIRO)	Todos os argumentos são VERDADEIRO (VERDADEIRO)
=E(VERDADEIRO; FALSO)	Um argumento é FALSO (FALSO)
=E(2+2=4; 2+3=5)	Todos os argumentos são avaliados como VERDADEIRO (VERDADEIRO)
=E(10>20;50<100)	Um argumento é FALSO (FALSO)

### NÃO

Inverte o valor do argumento. Use NÃO quando quiser ter certeza de que um valor não é igual a outro valor determinado.

#### Sintaxe

NÃO(lógico)

Lógico é um valor ou expressão que pode ser avaliado como VERDADEIRO ou FALSO.

#### Comentários

Se lógico for FALSO, NÃO retornará VERDADEIRO; se lógico for VERDADEIRO, NÃO retornará FALSO.

Fórmula	Descrição (resultado)
=NÃO(FALSO)	Reverte FALSO (VERDADEIRO)
=NÃO(1+1=2)	Reverte uma equação que é avaliada como VERDADEIRO (FALSO)

### OU

Retorna VERDADEIRO se qualquer argumento for VERDADEIRO; retorna FALSO se todos os argumentos forem FALSOS.

#### Sintaxe

## **OU(lógico1;lógico2;...)**

Lógico1;lógico2,... são de uma a 30 condições que você deseja testar e que podem resultar em VERDADEIRO ou FALSO.

### **Comentários**

- Os argumentos devem ser avaliados como valores lógicos, como VERDADEIRO ou FALSO, ou em [matrizes](#) ou referências que contenham valores lógicos.
- Se um argumento de uma matriz ou referência contiver texto ou células vazias, esses valores serão ignorados.
- Se o intervalo especificado não contiver valores lógicos, OU retornará o valor de erro #VALOR!.
- Você pode usar uma fórmula de matriz OU para verificar se um valor ocorre em uma matriz. Para inserir uma fórmula de matriz, pressione CTRL+SHIFT+ENTER.

<b>Fórmula</b>	<b>Descrição (resultado)</b>
=OU(VERDADEIRO)	Um argumento é VERDADEIRO (VERDADEIRO)
=OU(1+1=1;2+2=5)	Todos os argumentos são avaliados como FALSO (FALSO)
=OU(VERDADEIRO;FALSO;VERDADEIRO)	Pelo menos um argumento é VERDADEIRO (VERDADEIRO)

## **SE**

Retorna um valor se uma condição que você especificou avaliar como VERDADEIRO e um outro valor se for avaliado como FALSO.

Use SE para conduzir testes condicionais sobre valores e fórmulas.

### **Sintaxe**

**SE(teste\_lógico;valor\_se\_verdadeiro;valor\_se\_falso)**

Teste\_lógico é qualquer valor ou expressão que possa ser avaliado como VERDADEIRO ou FALSO. Por exemplo, A10=100 é uma expressão lógica; se o valor da célula A10 for igual a 100, a expressão será considerada VERDADEIRO. Caso contrário, a expressão será considerada FALSO. Esse argumento pode usar qualquer operador de cálculo de comparação.

Valor\_se\_verdadeiro é o valor retornado se teste\_lógico for VERDADEIRO. Por exemplo, se esse argumento for a seqüência de caracteres de texto "Dentro do orçamento" e o argumento teste\_lógico for considerado VERDADEIRO, a função SE exibirá o texto "Dentro do orçamento". Se teste\_lógico for VERDADEIRO e valor\_se\_verdadeiro for vazio, o argumento retornará 0 (zero). Para exibir a palavra VERDADEIRO, use o valor lógico VERDADEIRO para esse argumento. Valor\_se\_verdadeiro pode ser outra fórmula.

Valor\_se\_falso é o valor retornado se teste\_lógico for FALSO. Por exemplo, se esse argumento for a seqüência de caracteres de texto "Acima do orçamento" e o argumento teste\_lógico for considerado FALSO, a função SE exibirá o texto "Acima do orçamento". Se teste\_lógico for FALSO e valor\_se\_falso for omitido (ou seja, se não houver vírgula após valor\_se\_verdadeiro), o valor lógico FALSO será retornado. Se teste\_lógico for FALSO e valor\_se\_falso for vazio (ou seja, se houver uma vírgula após valor\_se\_verdadeiro seguida do parênteses de fechamento), o valor 0 (zero) será retornado. Valor\_se\_falso pode ser outra fórmula.

### **Comentários**

- É possível aninhar até sete funções SE como argumentos valor\_se\_verdadeiro e valor\_se\_falso para construir testes mais elaborados. Consulte o último dos exemplos a seguir.
- Quando os argumentos valor\_se\_verdadeiro e valor\_se\_falso são avaliados, SE retorna o valor que foi retornado por estas instruções.
- O Microsoft Excel fornece funções adicionais que podem ser usadas para analisar os dados com base em uma condição. Por exemplo, para contar o número de ocorrências de uma seqüência de caracteres de texto ou um número dentro de um intervalo de células, use a função de planilha CONT.SE. Para calcular uma soma baseada em uma seqüência de caracteres de texto ou em um número dentro de um intervalo, use a função de planilha SOMASE. Saiba como calcular um valor baseado em uma condição.

**Exemplo 1**

	A
1	Dados
2	50

Fórmula	Descrição (resultado)
=SE(A2<=100;"Dentro do orçamento";"Acima do orçamento")	Se o número acima for menor ou igual a 100, a fórmula exibirá "Dentro do orçamento". Caso contrário, a função exibirá "Acima do orçamento" (Dentro do orçamento)
=SE(A2=100;SOMA(B5:B15);"")	Se o número acima for 100, o intervalo B5:B15 será calculado. Caso contrário, o texto vazio ("") será retornado ( )

**Exemplo 2**

	A	B
	Despesas reais	Despesas previstas
1		
2	1500	900
3	500	900
4	500	925

Fórmula	Descrição (resultado)
=SE(A2>B2;"Acima do orçamento";"OK")	do Verifica se a primeira linha está acima do orçamento (Acima do orçamento)
=SE(A3>B3;"Acima do orçamento";"OK")	do Verifica se a segunda linha está acima do orçamento (OK)

**Exemplo 3**

	A
1	Resultado
2	45
3	90
4	78

Fórmula	Descrição (resultado)
=SE(A2>89;"A";SE(A2>79;"B";SE(A2>69;"C";SE(A2>59;"D";"F"))))	Atribui uma letra ao primeiro resultado (F)
=SE(A3>89;"A";SE(A3>79;"B";SE(A3>69;"C";SE(A3>59;"D";"F"))))	Atribui uma letra ao segundo resultado (A)
=SE(A4>89;"A";SE(A4>79;"B";SE(A4>69;"C";SE(A4>59;"D";"F"))))	Atribui uma letra ao terceiro resultado (C)

No exemplo anterior, a segunda instrução SE também é o argumento valor\_se\_falso para a primeira instrução SE. Da mesma maneira, a terceira instrução SE é o argumento valor\_se\_falso para a segunda instrução SE. Por exemplo, se o primeiro teste\_lógico (Média>89) for VERDADEIRO, "A" será retornado. Se o primeiro teste\_lógico for FALSO, a segunda instrução SE é avaliada e assim por diante.

As letras são atribuídas a números usando a seguinte chave.

Se o resultado for	Então retornará
Maior do que 89	A
De 80 a 89	B
De 70 a 79	C
De 60 a 69	D
Menor do que 60	F

**FUNÇÕES DE INFORMAÇÃO**
**CÉL**

Retorna informações sobre formatação, localização ou conteúdo da célula superior esquerda em uma referência.

**Sintaxe**

**CÉL(tipo\_info;ref)**

Tipo\_info é um valor de texto que especifica o tipo de informações da célula desejada. A lista a seguir apresenta os possíveis valores para tipo\_info e os resultados correspondentes.

Tipo_info	Retorna
"arquivo"	Nome do arquivo (incluindo nome completo do caminho) para o arquivo que contém referência, como texto. Retorna o texto vazio ("") se a planilha que contém referência ainda não tiver sido salva.
"col"	Número de coluna da célula em referência.
"conteúdo"	Valor da célula superior esquerda na referência, não uma fórmula.
"cor"	1 se a célula for formatada em cores para valores negativos; caso contrário retorna 0.
"endereço"	Referência da primeira célula em referência, como texto.
"formato"	Valor do texto correspondente ao formato de número da célula. Os valores de texto para os vários formatos são apresentados na tabela a seguir. Retorna "-" ao final do valor do texto se a célula for formatada em cores para valores negativos. Retorna "(" ao final do valor do texto se a célula for formatada com parênteses para os valores positivos ou todos os valores.
"largura"	A largura de coluna da célula arredondada para um inteiro. Cada unidade de largura de coluna é igual à largura de um caractere do tamanho de fonte padrão.
"linha"	Número de linha da célula em referência.
"parênteses"	1 se a célula for formatada com parênteses para os valores positivos ou todos os valores; caso contrário retorna 0.
"prefixo"	O valor do texto correspondente ao "prefixo de título" da célula. Retorna apóstrofo (') se a célula contiver texto alinhado à esquerda, aspas (") se a célula contiver texto alinhado à direita, acento circunflexo (^) se a célula contiver texto centralizado, barra invertida (\) se a célula contiver texto alinhado por preenchimento, e texto vazio ("") para outros conteúdos.
"proteção"	0 se a célula não for travada, e 1 se a célula for travada.
"tipo"	Valor de texto correspondente ao tipo de dados na célula. Retorna "b" se a célula estiver em branco, "l" se a célula contiver uma constante de texto, e "v" para outros conteúdos.

Ref é a célula sobre a qual você deseja obter informações. Se for omitida, as informações especificadas em tipo\_info serão retornadas com base na última célula alterada. A lista a seguir descreve os valores de texto que CÉL retorna quando tipo\_info for "formato" e ref for uma célula formatada com um formato de número interno.

Se o formato do Microsoft Excel for	CÉL retornará
Geral	"G"
0	"F0"
###0	".0"
0,00	"F2"
###0,00	".2"
R\$ ###0_);(R\$ ###0)	"C0"
R\$ ###0_);[Vermelho](R\$ ###0)	"C0-"
R\$ ###0,00_);(R\$ ###0,00)	"C2"
R\$ ###0,00_);[Vermelho](R\$ ###0,00)	"C2-"
0%	"P0"
0,00%	"P2"
0,00E+00	"S2"
# ?/? ou # ??/??	"G"
d/m/aa ou d/m/aa h:mm ou dd/mm/aa	"D4"
d-mmm-aa ou dd-mmm-aa	"D1"
d-mmm ou dd-mmm	"D2"
mmm-aa	"D3"
dd/mm	"D5"
h:mm AM/PM	"D7"
h:mm:ss AM/PM	"D6"

h:mm	"D9"
h:mm:ss	"D8"

Se o argumento tipo\_info na fórmula de CÉL for "formato" e a célula for posteriormente formatada com um formato personalizado, você deverá recalcular a planilha para atualizar a fórmula de CÉL.

### Comentários

A função CÉL é fornecida para compatibilidade com outros programas de planilha eletrônica.

	A
1	Dados
2	5/mar
3	TOTAL

Fórmula	Descrição (resultado)
=CÉL("linha";A20)	O número da linha da célula A20 (20)
=CÉL("formato"; A2)	O código de formato da primeira seqüência de caracteres (D2, veja acima)
=CÉL("conteúdo"; A3)	O conteúdo da célula A3 (TOTAL)

### Funções É

Esta seção descreve as nove funções de planilha usadas para testar o tipo de um valor ou referência.

Cada uma destas funções, chamadas coletivamente de funções É, verifica o tipo de valor e retorna VERDADEIRO ou FALSO, dependendo do resultado. Por exemplo, a função ÉCÉL.VAZIA retorna o valor lógico VERDADEIRO se o valor for uma referência a uma célula em branco; caso contrário, retorna FALSO.

#### Sintaxe

ÉCÉL.VAZIA(valor)

ÉERRO(valor)

ÉERROS(valor)

ÉLÓGICO(valor)

É.NÃO.DISP(valor)

É.NÃO.TEXTO(valor)

ÉNÚM(valor)

ÉREF(valor)

ÉTEXTO(valor)

Valor é o valor que se deseja testar. O valor pode ser um espaço em branco (célula vazia), um valor de erro, lógico, texto, número ou referência, ou um nome correspondente a qualquer um destes itens que se deseja testar.

Função	Retorna VERDADEIRO se
É.NÃO.DISP	Valor se referir ao valor de erro #N/D (valor não disponível).
É.NÃO.TEXTO	Valor se referir a qualquer item que não seja texto. Observe que esta função retorna VERDADEIRO se o valor se referir a uma célula em branco.
ÉCÉL.VAZIA	Valor se referir a uma célula vazia.
ÉERRO	Valor se referir a um valor de erro exceto #N/D.
ÉERROS	Valor se referir a qualquer valor de erro (#N/D, #VALOR!, #REF!, #DIV/0!, #NÚM!, #NOME? ou #NULO!).
ÉLÓGICO	Valor se referir a um valor lógico.
ÉNÚM	Valor se referir a um número.
ÉREF	Valor se referir a uma referência.
ÉTEXTO	Valor se referir a texto.

## Comentários

- Os argumentos de valor das funções É não são convertidos. Por exemplo, na maioria das outras funções em que se exige um número, o valor de texto "19" é convertido para o número 19. No entanto, na fórmula ÉNÚM("19"), "19" não é convertido de um valor de texto e a função ÉNÚM retorna FALSO.
- As funções É são úteis nas fórmulas para testar o resultado de um cálculo. Quando combinadas com a função SE, fornecem um método para localizar erros em fórmulas (consulte os exemplos a seguir).

### Exemplo 1

Fórmula	Descrição (resultado)
=ÉLÓGICO(VERDADEIRO)	Verifica se VERDADEIRO é um valor lógico (VERDADEIRO)
=ÉLÓGICO("VERDADEIRO")	Verifica se "VERDADEIRO" é um valor lógico (FALSO)
=ÉNÚM(4)	Verifica se 4 é um número (VERDADEIRO)

### Exemplo 2

	A	Fórmula	Descrição (resultado)
1	Dados	=ÉCÉL.VAZIA(A2)	Verifica se a célula C2 está vazia (FALSO)
2	Ouro	=ÉERROS(A4)	Verifica se #REF! é um erro (VERDADEIRO)
3	Região1	=É.NÃO.DISP(A4)	Verifica se #REF! é o erro #N/D (FALSO)
4	#REF!	=É.NÃO.DISP(A6)	Verifica se #N/D é o erro #N/D (VERDADEIRO)
5	330,92	=ÉERRO(A6)	Verifica se #N/D é um erro (FALSO)
6	#N/D	=ÉNÚM(A5)	Verifica se 330,92 é um número (VERDADEIRO)
		=ÉTEXTO(A3)	Verifica se Região1 é texto (VERDADEIRO)

## INFORMAÇÃO

**Segurança** Tenha cuidado ao usar este recurso. Informações sigilosas ou confidenciais podem ser reveladas a outros usuários.

Retorna informações sobre o ambiente operacional atual.

### Sintaxe

**INFORMAÇÃO(tipo\_texto)**

Tipo\_texto é o texto que especifica o tipo de informação a ser retornado.

Tipo_texto	Retorna
"diretório"	Caminho do diretório ou pasta atual.
"memdisp"	Quantidade de memória disponível em bytes.
"memtot"	Memória total disponível, incluindo a memória já utilizada, em bytes.
"memusada"	Quantidade de memória sendo utilizada para dados.
"númarquivo"	O número de planilhas ativas nas pastas de trabalho abertas.
"origem"	Referência absoluta do estilo A1, como texto, precedida de "\$A:" para compatibilidade com o Lotus 1-2-3 versão 3.x. Retorna a referência da célula superior da esquerda visível na janela com base na posição de rolagem atual.
"osversão"	Versão do sistema operacional atual na forma de texto.
"recalc"	Modo atual de refazer o cálculo; retorna "Automático" ou "Manual".
"release"	Versão do Microsoft Excel na forma de texto.
"sistema"	Nome do ambiente operacional: Macintosh = "mac" Windows = "pcdos"

Fórmula	Descrição (resultado)
=INFORMAÇÃO("númerarquivo")	O número de planilhas ativas (varia)
=INFORMAÇÃO("recalc")	O modo de refazer o cálculo da pasta de trabalho (Automático ou Manual)

## FUNÇÕES FINANCEIRAS

### **NP**ER

Retorna o número de períodos para investimento de acordo com pagamentos constantes e periódicos e uma taxa de juros constante.

#### Sintaxe

**NP**ER(taxa;pgto;vp;vf;tipo)

Para obter uma descrição completa dos argumentos em NP

ER e sobre as funções de anuidade, consulte VP.

Taxa é a taxa de juros por período.

Pgto é o pagamento feito em cada período; não pode mudar durante a vigência da anuidade. Geralmente, pgto contém o capital e os juros, mas nenhuma outra tarifa ou taxas.

Vp é o valor presente ou atual de uma série de pagamentos futuros.

Vf é o valor futuro, ou o saldo, que você deseja obter depois do último pagamento. Se vf for omitido, será considerado 0 (o valor futuro de um empréstimo, por exemplo, é 0).

Tipo é o número 0 ou 1 e indica as datas de vencimento.

Definir tipo para	Se os vencimentos forem
0 ou omitido	No final do período
1	No início do período

Exemplo:

Quanto tempo será necessário para juntar R\$ 5.000,00, poupando R\$ 250,00 por mês, a uma taxa de juros de 3,3% ao mês?

=NP

ER(3,33%;250;-5000) resulta em 33,48

### **PG**TO

Retorna o pagamento periódico de uma anuidade de acordo com pagamentos constantes e com uma taxa de juros constante.

#### Sintaxe

**PG**TO(taxa;nper;vp;vf;tipo)

Para obter uma descrição mais completa dos argumentos em PG

TO, consulte a função VP.

Taxa é a taxa de juros por período.

Nper é o número total de pagamentos pelo empréstimo.

Vp é o valor presente—o valor total presente de uma série de pagamentos futuros.

Vf é o valor futuro, ou o saldo de caixa, que você deseja obter depois do último pagamento. Se vf for omitido, será considerado 0 (o valor futuro de determinado empréstimo, por exemplo, 0).

Tipo é o número 0 ou 1 e indica as datas de vencimento.

Definir tipo para	Se os vencimentos forem
0 ou omitido	No final do período
1	No início do período

### Comentários

- O pagamento retornado por PGTO inclui o principal e os juros e não inclui taxas, pagamentos de reserva ou tarifas, às vezes associados a empréstimos.
- Certifique-se de que esteja sendo consistente quanto às unidades usadas para especificar taxa e nper. Se fizer pagamentos mensais por um empréstimo de quatro anos com juros de 12% ao ano, utilize 12%/12 para taxa e 4\*12 para nper. Se fizer pagamentos anuais para o mesmo empréstimo, use 12% para taxa e 4 para nper.

Exemplo:

Qual o valor da prestação que se pagará para juntar R\$ 8000,00 em 30 meses, a uma taxa de juros de 3,6% ao mês?

=PGTO(3,6%;30;-8000) resulta em 440,44

### TAXA

Retorna a taxa de juros por período de uma anuidade. TAXA é calculado por iteração e pode ter zero ou mais soluções. Se os resultados sucessivos de TAXA não convergirem para 0,0000001 depois de 20 iterações, TAXA retornará o valor de erro #NÚM!.

#### Sintaxe

**TAXA(nper;pgto;vp;vf;tipo;estimativa)**

Consulte VP para obter uma descrição completa dos argumentos nper, pgto, vp, vf e tipo.

Nper é o número total de períodos de pagamento em uma anuidade.

Pgto é o pagamento feito em cada período e não pode mudar durante a vigência da anuidade. Geralmente, pgto inclui o principal e os juros e nenhuma outra taxa ou tributo. Se pgto for omitido, você deverá incluir o argumento vf.

Vp é o valor presente — o valor total correspondente ao valor atual de uma série de pagamentos futuros.

Vf é o valor futuro, ou o saldo, que você deseja obter depois do último pagamento. Se vf for omitido, será considerado 0 (o valor futuro de um empréstimo, por exemplo, é 0).

Tipo é o número 0 ou 1 e indica as datas de vencimento.

Definir tipo para	Se os vencimentos forem
0 ou omitido	No final do período
1	No início do período

Estimativa é a sua estimativa para a taxa.

- Se você omitir estimativa, este argumento será considerado 10%.
- Se TAXA não convergir, atribua valores diferentes para estimativa. Em geral, TAXA converge se estimativa estiver entre 0 e 1.

### Comentários

Certifique-se de que esteja sendo consistente quanto às unidades usadas para especificar estimativa e nper. Se você fizer pagamentos mensais por um empréstimo de quatro anos com juros de 12% ao ano, utilize 12%/12 para estimativa e 4\*12 para nper. Se fizer pagamentos anuais para o mesmo empréstimo, utilize 12% para estimativa e 4 para nper.

Exemplo:

Qual a taxa de juros de um financiamento em que o valor à vista é de R\$ 3000,00, e vai ser pago em 15 parcelas de R\$340,00?

=TAXA(15;340;-3000) resulta em 7,5%

## VF

Retorna o valor futuro de um investimento de acordo com os pagamentos periódicos e constantes e com uma taxa de juros constante.

### Sintaxe

**VF(taxa;nper;pgto;vp;tipo)**

Para obter uma descrição completa dos argumentos em VF e para obter mais informações sobre as funções de anuidade, consulte VP.

Taxa é a taxa de juros por período.

Nper é o número total de períodos de pagamento em uma anuidade.

Pgto é o pagamento feito a cada período; não pode mudar durante a vigência da anuidade. Geralmente, pgto contém o capital e os juros e nenhuma outra tarifa ou taxas. Se pgto for omitido, você deverá incluir o argumento vp.

Vp é o valor presente ou a soma total correspondente ao valor presente de uma série de pagamentos futuros. Se vp for omitido, será considerado 0 (zero) e a inclusão do argumento pgto será obrigatória.

Tipo é o número 0 ou 1 e indica as datas de vencimento dos pagamentos. Se tipo for omitido, será considerado 0.

Definir tipo para	Se os vencimentos forem
0	No final do período
1	No início do período

### Comentários

- Certifique-se de que esteja sendo consistente quanto às unidades usadas para especificar taxa e nper. Se fizer pagamentos mensais de um empréstimo de quatro anos com taxa de juros de 12% ao ano, use 12%/12 para taxa e 4\*12 para nper. Se você fizer pagamentos anuais para o mesmo empréstimo, use 12% para taxa e 4 para nper.
- Todos os argumentos, saques, tais como depósitos em poupança, serão representados por números negativos; depósitos recebidos, tais como cheques de dividendos, serão representados por números positivos.

Exemplo:

Deseja-se fazer uma poupança de R\$ 300,00 por mês, durante 20 meses. A taxa de juros mensal é de 2,5% ao mês. Qual o valor total poupado no fim do período?

=VF(2,5%;20;-300) resulta em 7663,40.

## VP

Retorna o valor presente de um investimento. O valor presente é o valor total correspondente ao valor atual de uma série de pagamentos futuros. Por exemplo, quando você pede dinheiro emprestado, o valor do empréstimo é o valor presente para quem empresta.

### Sintaxe

**VP(taxa;nper;pgto;vf;tipo)**

Taxa é a taxa de juros por período. Por exemplo, se você obtiver um empréstimo para um carro com uma taxa de juros de 10% ao ano e fizer pagamentos mensais, a sua taxa de juros mensal será 10%/12, ou 0,83%. Você deve inserir 10%/12, ou 0,83%, ou 0,0083, na fórmula como taxa.

Nper é o número total de períodos de pagamento de uma anuidade. Por exemplo, se você obtiver um empréstimo de quatro anos e fizer pagamentos mensais, o empréstimo terá 4\*12 (ou 48) períodos. Você deve inserir 48 na fórmula para nper.

Pgto é o pagamento feito a cada período e não pode mudar durante a vigência da anuidade. Geralmente, pgto inclui o principal e os juros, e não há outras tarifas ou taxas. Por exemplo, os pagamentos mensais por um empréstimo para o carro de R\$ 10.000 de quatro anos a 12% são R\$ 263,33. Você deve inserir -263,33 na fórmula como pgto. Se pgto for omitido, você deverá incluir o argumento vf.

Vf é o valor futuro, ou um saldo de caixa, que você deseja obter depois do último pagamento. Se vf for omitido, será considerado 0 (o valor futuro de determinado empréstimo, por exemplo, é 0). Por exemplo, se quiser economizar R\$ 50.000 para pagar um projeto especial em 18 anos, então R\$ 50.000 é o valor futuro. Você pode então calcular a taxa de juros e determinar quanto deverá economizar a cada mês. Se vf for omitido, você deverá incluir o argumento pgto.

Tipo é o número 0 ou 1 e indica as datas de vencimento.

Definir tipo para	Se os vencimentos forem
0 ou omitido	No final do período
1	No início do período

### Comentários

- Certifique-se de que esteja sendo consistente quanto às unidades usadas para especificar taxa e nper. Se fizer pagamentos mensais de um empréstimo de quatro anos com taxa de juros de 12% ao ano, use 12%/12 para taxa e 4\*12 para nper. Se você fizer pagamentos anuais para o mesmo empréstimo, use 12% para taxa e 4 para nper.

Exemplo:

Uma loja oferece um refrigerador em 10 parcelas de R\$ 120,00. A loja informa que a taxa de juros é de 4,5% ao mês. Qual o valor à vista do produto?

=VP(4,5%;10;-120) resulta em 949,53.